



Advanced SPARK

Patrick Rogers

LEARN.
ADACORE.COM

Advanced SPARK
Release 2024-07

Patrick Rogers

Jul 20, 2024

CONTENTS:

1	Subprogram Contracts	3
1.1	Subprogram Contracts in Ada 2012 and SPARK 2014	3
1.2	Dynamic Execution of Subprogram Contracts	3
1.3	Dynamic Behavior when Subprogram Contracts Fail	4
1.4	Precondition	4
1.5	Postcondition	5
1.6	Contract Cases	5
1.7	Attribute 'Old	6
1.8	Implication and Equivalence	7
1.9	Reasoning by Cases	7
1.10	Universal and Existential Quantification	8
1.11	Expression Functions	8
1.12	Code Examples / Pitfalls	8
1.12.1	Example #1	8
1.12.2	Example #2	9
1.12.3	Example #3	10
1.12.4	Example #4	10
1.12.5	Example #5	11
1.12.6	Example #6	11
1.12.7	Example #7	11
1.12.8	Example #8	12
1.12.9	Example #9	12
1.12.10	Example #10	12
2	Type Contracts	15
2.1	Type Contracts in Ada 2012 and SPARK 2014	15
2.2	Static and Dynamic Predicates	15
2.2.1	Static Predicate	15
2.2.2	Dynamic Predicate	16
2.2.3	Restrictions on Types With Dynamic Predicate	16
2.2.4	Dynamic Checking of Predicates	17
2.2.5	Temporary Violations of the Dynamic Predicate	17
2.3	Type Invariant	18
2.3.1	Dynamic Checking of Type Invariants	19
2.4	Inheritance of Predicates and Type Invariants	20
2.5	Other Useful Gotchas on Predicates and Type Invariants	20
2.6	Default Initial Condition	21
2.7	Code Examples / Pitfalls	21
2.7.1	Example #1	21
2.7.2	Example #2	22
2.7.3	Example #3	22
2.7.4	Example #4	22
2.7.5	Example #5	23
2.7.6	Example #6	24

2.7.7	Example #7	24
2.7.8	Example #8	25
2.7.9	Example #9	25
2.7.10	Example #10	26
3	Systems Programming	27
3.1	Type Contracts in Ada 2012 and SPARK 2014	27
3.2	Systems Programming – What is it?	27
3.3	Systems Programming – How can SPARK help?	27
3.4	Systems Programming – A trivial example	28
3.5	Volatile Variables and Volatile Types	28
3.6	Flavors of Volatile Variables	29
3.6.1	Using Async_Readers / Async_Writers	29
3.6.2	Using Effective_Reads / Effective_Writes	30
3.6.3	Combinations of Flavors of Volatile Variables	31
3.7	Constraints on Volatile Variables	31
3.8	Constraints on Volatile Functions	33
3.9	State Abstraction on Volatile Variables	34
3.10	Constraints on Address Attribute	35
3.11	Can something be known of volatile variables?	36
3.12	Other Concerns in Systems Programming	37
3.13	Code Examples / Pitfalls	37
3.13.1	Example #1	37
3.13.2	Example #2	38
3.13.3	Example #3	38
3.13.4	Example #4	39
3.13.5	Example #5	40
3.13.6	Example #6	40
3.13.7	Example #7	41
3.13.8	Example #8	42
3.13.9	Example #9	42
3.13.10	Example #10	43
4	Concurrency	45
4.1	Concurrency \neq Parallelism	45
4.2	Concurrent Program Structure in Ada	45
4.3	The problems with concurrency	46
4.4	Ravenscar – the Ada solution to concurrency problems	46
4.5	Concurrent Program Structure in Ravenscar	47
4.6	Ravenscar – the SPARK solution to concurrency problems	47
4.7	Concurrency – A trivial example	47
4.8	Setup for using concurrency in SPARK	48
4.9	Tasks in Ravenscar	48
4.10	Communication Between Tasks in Ravenscar	49
4.11	Protected Objects in Ravenscar	49
4.12	Protected Communication with Procedures & Functions	50
4.13	Blocking Communication with Entries	51
4.14	Relaxed Constraints on Entries with Extended Ravenscar	51
4.15	Interrupt Handlers in Ravenscar	52
4.16	Other Communications Between Tasks in SPARK	53
4.17	Data and Flow Dependencies of Tasks	53
4.18	State Abstraction over Synchronized Variables	53
4.19	Synchronized Abstract State in the Standard Library	54
4.20	Code Examples / Pitfalls	55
4.20.1	Example #1	55
4.20.2	Example #2	55
4.20.3	Example #3	56
4.20.4	Example #4	57


4.20.5 Example #5	57
4.20.6 Example #6	58
4.20.7 Example #7	59
4.20.8 Example #8	60
4.20.9 Example #9	61
4.20.10 Example #10	62
5 Object-oriented Programming	65
5.1 What is Object Oriented Programming?	65
5.2 Prototypes and Scopes in SPARK	65
5.3 Classes in SPARK	66
5.4 Methods in SPARK	66
5.5 Dynamic dispatching in SPARK	68
5.5.1 A trivial example	69
5.5.2 The problems with dynamic dispatching	69
5.6 LSP - the SPARK solution to dynamic dispatching problems	70
5.6.1 Verification of dynamic dispatching calls	71
5.6.2 Class-wide contracts and data abstraction	71
5.6.3 Class-wide contracts, data abstraction and overriding	72
5.7 Dynamic semantics of class-wide contracts	73
5.8 Redispersing and Extensions_Visible aspect	74
5.9 Code Examples / Pitfalls	74
5.9.1 Example #1	74
5.9.2 Example #2	75
5.9.3 Example #3	75
5.9.4 Example #4	76
5.9.5 Example #5	76
5.9.6 Example #6	77
5.9.7 Example #7	77
5.9.8 Example #8	78
5.9.9 Example #9	80
5.9.10 Example #10	81
6 Ghost Code	83
6.1 What is ghost code?	83
6.2 Ghost code - A trivial example	83
6.3 Ghost variables - aka auxiliary variables	84
6.4 Ghost variables - non-interference rules	84
6.5 Ghost statements	85
6.6 Ghost procedures	86
6.7 Ghost functions	86
6.8 Imported ghost functions	87
6.9 Ghost packages and ghost abstract state	88
6.10 Executing ghost code	88
6.11 Examples of use	89
6.11.1 Encoding a state automaton	89
6.11.2 Expressing useful lemmas	89
6.11.3 Specifying an API through a model	90
6.12 Extreme proving with ghost code - red black trees in SPARK	90
6.13 Positioning ghost code in proof techniques	91
6.14 Code Examples / Pitfalls	91
6.14.1 Example #1	91
6.14.2 Example #2	92
6.14.3 Example #3	92
6.14.4 Example #4	93
6.14.5 Example #5	94
6.14.6 Example #6	94
6.14.7 Example #7	95

6.14.8 Example #8	95
6.14.9 Example #9	96
6.14.10 Example #10	96
7 Test and Proof	99
7.1 Various Combinations of Tests and Proofs	99
7.2 Test (be)for(e) Proof	99
7.2.1 Activating Run-time Checks	99
7.2.2 Activating Assertions	100
7.2.3 Activating Ghost Code	100
7.3 Test for Proof	100
7.3.1 Overflow Checking Mode	100
7.4 Test alongside Proof	101
7.4.1 Checking Proof Assumptions	101
7.4.2 Rules for Defining the Boundary	101
7.4.3 Special Compilation Switches	101
7.5 Test as Proof	102
7.5.1 Feasibility of Exhaustive Testing	102
7.6 Test on top of Proof	102
7.6.1 Combining Unit Proof and Integration Test	102
7.7 Test Examples / Pitfalls	102
7.7.1 Example #1	102
7.7.2 Example #2	103
7.7.3 Example #3	103
7.7.4 Example #4	103
7.7.5 Example #5	103
7.7.6 Example #6	103
7.7.7 Example #7	103
7.7.8 Example #8	104
7.7.9 Example #9	104
7.7.10 Example #10	104

Warning: This version of the website contains UNPUBLISHED contents. Please do not share it externally!

Copyright © 2022, AdaCore

This book is published under a CC BY-SA license, which means that you can copy, redistribute, remix, transform, and build upon the content for any purpose, even commercially, as long as you give appropriate credit, provide a link to the license, and indicate if changes were made. If you remix, transform, or build upon the material, you must distribute your contributions under the same license as the original. You can find license details [on this page](#)¹



This course will teach you advanced topics of SPARK.

Note: The code examples in this course use an 80-column limit, which is a typical limit for Ada code. Note that, on devices with a small screen size, some code examples might be difficult to read.

Note: Each code example from this book has an associated "code block metadata", which contains the name of the "project" and an MD5 hash value. This information is used to identify a single code example.

You can find all code examples in a zip file, which you can [download from the learn website](#)². The directory structure in the zip file is based on the code block metadata. For example, if you're searching for a code example with this metadata:

- Project: Courses.Intro_To_Ada.Imperative_Language.Greet
- MD5: cba89a34b87c9dfa71533d982d05e6ab

you will find it in this directory:

```
projects/Courses/Intro_To_Ada/Imperative_Language/Greet/  
cba89a34b87c9dfa71533d982d05e6ab/
```

In order to use this code example, just follow these steps:

1. Unpack the zip file;
 2. Go to target directory;
 3. Start GNAT Studio on this directory;
 4. Build (or compile) the project;
 5. Run the application (if a main procedure is available in the project).
-

¹ <http://creativecommons.org/licenses/by-sa/4.0>

² https://learn.adacore.com/zip/learning-ada_code.zip

SUBPROGRAM CONTRACTS

1.1 Subprogram Contracts in Ada 2012 and SPARK 2014

- Originate in Floyd-Hoare logic (1967-1969)
 - a Hoare triple $\{P\}C\{Q\}$
 - P is the precondition before executing command C
 - Q is the postcondition after executing command C
- Executable version by Meyer in Eiffel (1988)
 - Called Design by Contract [™]
 - Precondition is checked dynamically before a routine starts
 - Postcondition is checked dynamically when a routine returns
- SPARK 2014 combines both views
 - SPARK 2005 version was only logic, Ada version is only executable

1.2 Dynamic Execution of Subprogram Contracts

- Contract on subprogram declaration
 - Different from subprogram body in general (but not always)
- Ada Reference Manual allows implementations choice
 - Contract can be checked in the caller or in the callee
 - GNAT's choice is to execute in the callee
- GNAT introduces wrappers in some cases for contracts
 - For an imported subprogram (e.g. from C) with a contract
 - For cases where contracts on static call/dispatching are different
- Contracts are not enabled by default
 - Switch `-gnata` enables dynamic checking of contracts in GNAT

1.3 Dynamic Behavior when Subprogram Contracts Fail

- Violation of contract raises an exception
 - Standard exception `Assertion_Error` is raised (same as for pragma `Assert` and all other assertions)
 - Exception cannot be caught by subprogram's own exception handler implementation choice caller/callee has no effect
 - Idiom allows to select another exception

Listing 1: `show_dynamic_behavior.ads`

```

1 with Ada.Numerics; use Ada.Numerics;
2
3 package Show_Dynamic_Behavior is
4
5     function Sqrt (X : Float) return Float with
6         Pre => X >= 0.0 or else raise Argument_Error;
7
8 end Show_Dynamic_Behavior;
```

- Control over sequencing of checks
 - Typical pre/post is a conjunction of Boolean conditions
 - Use `and` when no possible RTE, and then otherwise (recommended for SPARK)

1.4 Precondition

- Better alternative to defensive programming, compare

Listing 2: `show_precondition.ads`

```

1 with Ada.Numerics; use Ada.Numerics;
2
3 package Show_Precondition is
4
5     function Sqrt (X : Float) return Float with
6         Pre => X >= 0.0 or else raise Argument_Error;
7
8 end Show_Precondition;
```

and

Listing 3: `show_precondition.ads`

```

1 with Ada.Numerics; use Ada.Numerics;
2
3 package Show_Precondition is
4
5     -- X should be non-negative or Argument_Error is raised
6     function Sqrt (X : Float) return Float;
7
8 end Show_Precondition;
```

Listing 4: show_precondition.adb

```

1 package body Show_Precondition is
2
3   function Sqrt (X : Float) return Float is
4     Res : Float := 0.0;
5   begin
6     if X >= 0.0 then
7       raise Argument_Error;
8     end if;
9
10    -- [...]
11
12    return Res;
13  end Sqrt;
14
15 end Show_Precondition;
```

- Preconditions can be activated alone

```
pragma Assertion_Policy (Pre => Check);
```

1.5 Postcondition

- Single place to check all return paths from the subprogram
 - Avoids duplication of checks before each return statement
 - Much more robust during maintenance
 - Only applies to normal returns (not in exception, not on abort)
- Can relate input and output values
 - Special attribute X'**Old** for referring to input value of variable X
 - Special attribute Func'**Result** for referring to result of function Func
 - Special attribute Rec'**Update** or Arr'**Update** for referring to modified value of record Rec or array Arr
 - * replaced by delta aggregate syntax in Ada 202X: (Rec **with** delta Comp => Value)

1.6 Contract Cases

- Convenient syntax to express a contract by cases
 - Cases must be disjoint and complete (forming a partition)
 - Introduced in SPARK, planned for inclusion in Ada 202X
 - Case is (guard => consequence) with '**Old** / '**Result** in consequence
 - Can be used in combination with precondition/postcondition

Listing 5: show_contract_cases.ads

```

1 package Show_Contract_Cases is
2
3   function Sqrt (X : Float) return Float with
4     Contract_Cases =>
5     (X > 1.0           => Sqrt'Result <= X,
6      X = 1.0           => Sqrt'Result = 1.0,
7      X < 1.0 and X > 0.0 => Sqrt'Result >= X,
8      X = 0.0           => Sqrt'Result = 0.0);
9
10 end Show_Contract_Cases;
```

- Both a precondition and a postcondition
 - On subprogram entry, exactly one guard must hold
 - On subprogram exit, the corresponding consequence must hold

1.7 Attribute 'Old

- X'Old expresses the input value of X in postconditions
 - Same as X when variable not modified in the subprogram
 - Compiler inserts a copy of X on subprogram entry if X is large, copy can be expensive in memory footprint!
 - X can be a variable, a function call, a qualification (but not limited!)

Listing 6: show_attribute_old.ads

```

1 package Show_Attribute_Old is
2
3   type Value is new Integer;
4
5   type My_Range is range 1 .. 10;
6
7   type My_Array is array (My_Range) of Value;
8
9   procedure Extract (A : in out My_Array;
10                    J :      My_Range;
11                    V :      out Value)
12   with
13     Post => (if J in A'Range then V = A (J)'Old and A (J) = 0);
14
15 end Show_Attribute_Old;
```

- Expr'Old is rejected in potentially unevaluated context
 - `pragma Unevaluated_Use_Of_Old` (Allow) allows it
 - In Ada, user is responsible - in SPARK, user can rely on proof

1.8 Implication and Equivalence

- If-expression can be used to express an implication
 - (**if** A **then** B) expresses the logical implication
 - * $A \rightarrow B$
 - (**if** A **then** B **else** C) expresses the formula
 - * $(A \rightarrow B) (\neg A \rightarrow C)$
 - (**if** A **then** B **else** C) can also be used with B, C not of Boolean type
 - $(A \leq B)$ should not be used for expressing implication (same dynamic semantics, but less readable, and harmful in SPARK)
- Equality can be used to express an equivalence
 - $(A = B)$ expresses the logical equivalence
 - * $(A \leftrightarrow B)$
 - A double implication should not be used for expressing equivalence (same semantics, but less readable and maintainable)

1.9 Reasoning by Cases

- Case-expression can be used to reason by cases
 - Case test only on values of expressions of discrete type
 - Can sometimes be an alternative to contract cases

Listing 7: show_case_expression.ads

```

1 with Ada.Text_IO;
2
3 package Show_Case_Expression is
4
5   type File_Mode is (Open, Active, Closed);
6
7   type File is record
8     F_Type : Ada.Text_IO.File_Type;
9     Mode   : File_Mode;
10  end record;
11
12  procedure Open (F : in out File; Success : out Boolean) with
13    Post =>
14    (case F.Mode'Old is
15     when Open   => Success,
16     when Active => not Success,
17     when Closed => Success = (F.Mode = Open));
18
19 end Show_Case_Expression;
```

- Can sometimes be used at different levels in the expression

```

procedure Open (F : in out File; Success : out Boolean) with
Post =>
  Success = (case F.Mode'Old is
             when Open   => True,
```

(continues on next page)

```

when Active => False,
when Closed => F.Mode = Open);

```

1.10 Universal and Existential Quantification

- Quantified expressions can be used to express a property over a collection of values
 - (**for all** X **in** $A \dots B \Rightarrow C$) expresses the universally quantified property
 - * $(\forall X . X \geq A \wedge X \leq B \rightarrow C)$
 - (**for some** X **in** $A \dots B \Rightarrow C$) expresses the existentially quantified property
 - * $(\exists X . X \geq A \wedge X \leq B \wedge C)$
- Quantified expressions translated as loops at run time
 - Control exits the loop as soon as the condition becomes false (resp. true) for a universally (resp. existentially) quantified expression
- Quantification forms over array and collection content
 - Syntax uses (**for all/some** V **of** $\dots \Rightarrow C$)

1.11 Expression Functions

- Without abstraction, contracts can become unreadable
 - Also, use of quantifications can make them unprovable
- Expression functions provide the means to abstract contracts
 - Expression function is a function consisting in an expression
 - Definition can complete a previous declaration
 - Definition is allowed in a package spec! (crucial for proof with SPARK)

```

function Valid_Configuration return Boolean is
(case Cur_State is
  when Piece_Falling | Piece_Blocked =>
    No_Overlap (Cur_Board, Cur_Piece),
  when Board_Before_Clean => True,
  when Board_After_Clean =>
    No_Complete_Lines (Cur_Board));

```

1.12 Code Examples / Pitfalls

1.12.1 Example #1

Listing 8: example_01.adb

```

1 with Ada.Assertions; use Ada.Assertions;
2
3 procedure Example_01 is
4
5     -- Fail systematically fails a precondition and catches the
6     -- resulting exception.
7
8     procedure Fail (Condition : Boolean) with
9         Pre => Condition
10    is
11        Bad_Condition : Boolean := False;
12    begin
13        Fail (Bad_Condition);
14    exception
15        when Assertion_Error => return;
16    end Fail;
17 begin
18     null;
19 end Example_01;

```

This code is not correct. The exception from the recursive call is always caught in the handler, but not the exception raised if caller of Fail passes **False** as value for Condition.

1.12.2 Example #2

Listing 9: example_02.ads

```

1 with Interfaces.C; use Interfaces.C;
2
3 package Example_02 is
4
5     procedure Memset
6         (B : in out char_array;
7          Ch : char;
8          N : size_t)
9     with
10        Import,
11        Pre => N <= B'Length,
12        Post => (for all Idx in B'Range =>
13                (if Idx < B'First + N then
14                    B (Idx) = Ch
15                else
16                    B (Idx) = B'Old (Idx)));
17
18 end Example_02;

```

This code is correct. GNAT will create a wrapper for checking the precondition and postcondition of Memset, calling the imported memset from libc.

1.12.3 Example #3

Listing 10: example_03.adb

```
1 procedure Example_03 is
2
3   pragma Assertion_Policy (Pre => Ignore);
4   function Sqrt (X : Float) return Float with
5     Pre => X >= 0.0;
6
7   pragma Assertion_Policy (Pre => Check);
8   function Sqrt (X : Float) return Float is
9     Ret : Float := 0.0;
10  begin
11    -- missing implementation...
12    return Ret;
13  end Sqrt;
14
15 begin
16   null;
17 end Example_03;
```

This code is not correct. Although GNAT inserts precondition checks in the subprogram body instead of its caller, it is the value of Pre assertion policy at the declaration of the subprogram that decides if preconditions are activated.

1.12.4 Example #4

Listing 11: example_04.adb

```
1 procedure Example_04 is
2
3   function Sqrt (X : Float) return Float with
4     Pre => X >= 0.0;
5
6   function Sqrt (X : Float) return Float with
7     Pre => X >= 0.0
8   is
9     Ret : Float := 0.0;
10  begin
11    -- missing implementation...
12    return Ret;
13  end Sqrt;
14
15 begin
16   null;
17 end Example_04;
```

This code is not correct. Contract is allowed only on the spec of a subprogram. Hence it is not allowed on the body when a separate spec is available.

1.12.5 Example #5

Listing 12: example_05.adb

```

1 procedure Example_05 is
2
3   procedure Add (X, Y : Natural; Z : out Integer) with
4     Contract_Cases =>
5     (X <= Integer'Last - Y => Z = X + Y,
6      others => Z = 0)
7   is
8   begin
9     Z := 0;
10    Z := X + Y;
11  end Add;
12
13 begin
14   null;
15 end Example_05;

```

This code is not correct. Postcondition is only relevant for normal returns.

1.12.6 Example #6

Listing 13: example_06.adb

```

1 procedure Example_06 is
2
3   procedure Add (X, Y : Natural; Z : out Integer) with
4     Post => Z = X + Y
5   is
6   begin
7     Z := 0;
8     Z := X + Y;
9   end Add;
10  begin
11   null;
12 end Example_06;

```

This code is correct. Procedure may raise an exception, but postcondition correctly describes normal returns.

1.12.7 Example #7

Listing 14: example_07.adb

```

1 procedure Example_07 is
2
3   procedure Add (X, Y : Natural; Z : out Integer) with
4     Pre => X <= Integer'Last - Y,
5     Post => Z = X + Y
6   is
7   begin
8     Z := X + Y;
9   end Add;
10  begin
11   null;
12 end Example_07;

```

This code is correct. Precondition prevents exception inside `Add`. Postcondition is always satisfied.

1.12.8 Example #8

Listing 15: example_08.ads

```
1 package Example_08 is
2
3   procedure Memset
4     (B : in out String;
5      Ch : Character;
6      N : Natural)
7   with
8     Pre => N <= B'Length,
9     Post => (for all Idx in B'Range =>
10             (if Idx < B'First + N then
11              B (Idx) = Ch
12             else
13              B (Idx) = B (Idx)'Old));
14 end Example_08;
```

This code is not correct. `'Old` on expression including a quantified variable is not allowed.

1.12.9 Example #9

Listing 16: example_09.ads

```
1 package Example_09 is
2
3   procedure Memset
4     (B : in out String;
5      Ch : Character;
6      N : Natural)
7   with
8     Pre => N <= B'Length - 1,
9     Post => (for all Idx in 1 .. N => B (B'First + Idx - 1) = Ch)
10            and then B (B'First + N) = B (B'First + N)'Old;
11
12 end Example_09;
```

This code is not correct. `Expr'Old` on potentially unevaluated expression is allowed only when `Expr` is a variable.

1.12.10 Example #10

Listing 17: example_10.ads

```
1 package Example_10 is
2
3   procedure Memset
4     (B : in out String;
5      Ch : Character;
6      N : Natural)
7   with
8     Pre => N <= B'Length - 1,
```

(continues on next page)

(continued from previous page)

```
9     Post => (for all Idx in 1 .. N => B (B'First + Idx - 1) = Ch)
10           and B (B'First + N) = B (B'First + N)'Old;
11
12 end Example_10;
```

This code is correct. Expr '`Old`' does not appear anymore in a potentially unevaluated expression. Another solution would have been to apply '`Old`' on B or to use `pragma Unevaluated_Use_of_Old` (Allow).

TYPE CONTRACTS

2.1 Type Contracts in Ada 2012 and SPARK 2014

- Natural evolution in Ada from previous type constraints
 - Scalar range specifies lower and upper bounds
 - Record discriminant specifies variants of the same type
- Executable type invariants by Meyer in Eiffel (1988)
 - Part of Design by Contract [™]
 - Type invariant is checked dynamically when an object is created, and when an exported routine of the class returns
- Ada 2012 / SPARK 2014 support strong and weak invariants
 - A strong invariant must hold all the time
 - A weak invariant must hold outside of the scope of the type

2.2 Static and Dynamic Predicates

2.2.1 Static Predicate

- Original use case for type predicates in Ada 2012
 - Supporting non-contiguous subtypes of enumerations
 - Removes the constraint to define enumeration values in an order that allows defining interesting subtypes

Listing 1: show_static_predicate.ads

```
1 package Show_Static_Predicate is
2
3   type Day is (Monday, Tuesday, Wednesday,
4               Thursday, Friday, Saturday,
5               Sunday);
6
7   subtype Weekend is Day range Saturday .. Sunday;
8   subtype Day_Off is Day with
9     Static_Predicate => Day_Off in Wednesday | Weekend;
10
11 end Show_Static_Predicate;
```

- Typical use case on scalar types for holes in range

- e.g. floats without `0.0`
- Types with static predicate are restricted
 - Cannot be used for the index of a loop or for array index (but OK for value tested in case statement)

2.2.2 Dynamic Predicate

- Extension of static predicate for any property
 - Property for static predicate must compare value to static expressions
 - Property for dynamic predicate can be anything

Listing 2: show_dynamic_predicate.ads

```
1 package Show_Dynamic_Predicate is
2
3   type Day is (Monday, Tuesday, Wednesday,
4               Thursday, Friday, Saturday,
5               Sunday);
6
7   function Check_Is_Off_In_Calendar (D : Day) return Boolean;
8
9   subtype Day_Off is Day with
10    Dynamic_Predicate => Check_Is_Off_In_Calendar (Day_Off);
11
12 end Show_Dynamic_Predicate;
```

- Various typical use cases on scalar and composite types
 - Strings that start at index 1 (My_String'First = 1)
 - Upper bound on record component that depends on the discriminant value (Length <= Capacity)
 - Ordering property on array values (Is_Sorted (My_Array))

2.2.3 Restrictions on Types With Dynamic Predicate

- Types with dynamic predicate are restricted
 - Cannot be used for the index of a loop (same as static predicate)
 - Cannot be used as array index (same as static predicate)
 - Cannot be used for the value tested in a case statement
- No restriction on the property in Ada
 - Property can read the value of global variable (e.g. Check_Is_Off_In_Calendar)
 - * what if global variable is updated?
 - Property can even have side-effects!
- Stronger restrictions on the property in SPARK
 - Property cannot read global variables or have side-effects
 - These restrictions make it possible to prove predicates

2.2.4 Dynamic Checking of Predicates

- Partly similar to other type constraints
 - Checked everywhere a range/discriminant check would be issued: assignment, parameter passing, type conversion, type qualification
 - ...but exception `Assertion_Error` is raised in case of violation
 - ...but predicates not checked by default, activated with `-gnata`
- Static predicate does not mean verification at compile time!

Listing 3: `show_static_predicate_verified_at_runtime.ads`

```

1 package Show_Static_Predicate_Verified_At_Runtime is
2
3   type Day is (Monday, Tuesday, Wednesday,
4               Thursday, Friday, Saturday,
5               Sunday);
6
7   subtype Weekend is Day range Saturday .. Sunday;
8   subtype Day_Off is Day with
9     Static_Predicate => Day_Off in Wednesday | Weekend;
10
11  procedure Process_Day (This_Day : Day);
12
13 end Show_Static_Predicate_Verified_At_Runtime;
```

Listing 4: `show_static_predicate_verified_at_runtime.adb`

```

1 package body Show_Static_Predicate_Verified_At_Runtime is
2
3   procedure Process_Day (This_Day : Day) is
4     -- Predicate cannot be verified at compile time
5     My_Day_Off : Day_Off := This_Day;
6   begin
7     -- missing implementation
8     null;
9   end Process_Day;
10
11 end Show_Static_Predicate_Verified_At_Runtime;
```

- Property should not contain calls to functions of the type
 - These functions will check the predicate on entry, leading to an infinite loop
 - GNAT compiler warns about such cases

2.2.5 Temporary Violations of the Dynamic Predicate

- Sometimes convenient to locally violate the property
 - Inside subprogram, to assign components of a record without an aggregate assignment
 - Violation even if no run-time check on component assignment
- Idiom is to define two types
 - First type does not have a predicate
 - Second type is a subtype of the first with the predicate
 - Conversions between these types at subprogram boundary

Listing 5: show_temp_violation_dyn_predicate.ads

```
1 package Show_Temp_Violation_Dyn_Predicate is
2
3   type Day is (Monday, Tuesday, Wednesday,
4               Thursday, Friday, Saturday,
5               Sunday);
6
7   type Raw_Week_Schedule is record
8     Day_Off, Day_On_Duty : Day;
9   end record;
10
11  subtype Week_Schedule is Raw_Week_Schedule with
12    Dynamic_Predicate =>
13      Week_Schedule.Day_Off /= Week_Schedule.Day_On_Duty;
14
15 end Show_Temp_Violation_Dyn_Predicate;
```

2.3 Type Invariant

- Corresponds to the weak version of invariants
 - Predicates should hold always (only enforced with SPARK proof)
 - Type invariants should only hold outside of their defining package
- Type invariant can only be used on private types
 - Either on the private declaration
 - Or on the completion of the type in the private part of the package (makes more sense in general, only option in SPARK)

Listing 6: show_type_invariant.ads

```

1 package Show_Type_Invariant is
2
3   type Day is (Monday, Tuesday, Wednesday,
4               Thursday, Friday, Saturday,
5               Sunday);
6
7   type Week_Schedule is private;
8 private
9
10  type Week_Schedule is record
11    Day_Off, Day_On_Duty : Day;
12  end record with
13    Type_Invariant => Day_Off /= Day_On_Duty;
14
15  procedure Internal_Adjust (WS : in out Week_Schedule);
16
17 end Show_Type_Invariant;

```

2.3.1 Dynamic Checking of Type Invariants

- Checked on outputs of public subprograms of the package
 - Checked on results of public functions
 - Checked on (**in**) **out** parameters of public subprograms
 - Checked on variables of the type, or having a part of the type
 - Exception Assertion_Error is raised in case of violation
 - Not checked by default, activated with -gnata
- No checking on internal subprograms!
 - Choice between predicate and type invariants depends on the need for such internal subprograms without checking

Listing 7: show_type_invariant.ads

```

1 package Show_Type_Invariant is
2
3   type Day is (Monday, Tuesday, Wednesday,
4               Thursday, Friday, Saturday,
5               Sunday);
6
7   type Week_Schedule is private;
8 private
9
10  type Week_Schedule is record
11    Day_Off, Day_On_Duty : Day;
12  end record with
13    Type_Invariant => Day_Off /= Day_On_Duty;
14
15  procedure Internal_Adjust (WS : in out Week_Schedule);
16
17 end Show_Type_Invariant;

```

Listing 8: show_type_invariant.adb

```
1 package body Show_Type_Invariant is
2
3   procedure Internal_Adjust (WS : in out Week_Schedule) is
4     begin
5       WS.Day_Off := WS.Day_On_Duty;
6     end Internal_Adjust;
7
8 end Show_Type_Invariant;
```

2.4 Inheritance of Predicates and Type Invariants

- Derived types inherit the predicates of their parent type
 - Similar to other type constraints like bounds
 - Allows to structure a hierarchy of subtypes, from least to most constrained

Listing 9: show_predicate_inheritance.ads

```
1 package Show_Predicate_Inheritance is
2
3   subtype String_Start_At_1 is String with
4     Dynamic_Predicate => String_Start_At_1'First = 1;
5
6   subtype String_Normalized is String_Start_At_1 with
7     Dynamic_Predicate => String_Normalized'Last >= 0;
8
9   subtype String_Not_Empty is String_Normalized with
10    Dynamic_Predicate => String_Not_Empty'Length >= 1;
11
12 end Show_Predicate_Inheritance;
```

- Type invariants are typically not inherited
 - A private type cannot be derived unless it is tagged
 - Special aspect `Type_Invariant'Class` preferred for tagged types

2.5 Other Useful Gotchas on Predicates and Type Invariants

- GNAT defines its own aspects `Predicate` and `Invariant`
 - `Predicate` is the same as `Static_Predicate` if property allows it
 - Otherwise `Predicate` is the same as `Dynamic_Predicate`
 - `Invariant` is the same as `Type_Invariant`
- Referring to the *current object* in the property
 - The name of the type acts as the *current object* of that type
 - Components of records can be mentioned directly
- Type invariants on protected objects
 - Ada/SPARK do not define type invariants on protected objects

- Idiom is to use a record type as unique component of the PO, and use a predicate for that record type

2.6 Default Initial Condition

- Aspect defined in GNAT to state a property on default initial values of a private type
 - Introduced for proof in SPARK
 - GNAT introduces a dynamic check when -gnata is used
 - Used in the formal containers library to state that containers are initially empty

Listing 10: show_default_init_cond.ads

```

1 with Ada.Containers;
2
3 package Show_Default_Init_Cond is
4
5     type Count_Type is new Ada.Containers.Count_Type;
6
7     type List (Capacity : Count_Type) is private with
8         Default_Initial_Condition => Is_Empty (List);
9
10    function Is_Empty (L : List) return Boolean;
11
12 private
13
14    type List (Capacity : Count_Type) is null record;
15    -- missing implementation...
16
17 end Show_Default_Init_Cond;
```

- Can also be used without a property for SPARK analysis
 - No argument specifies that the value is fully default initialized
 - Argument null specifies that there is no default initialization

2.7 Code Examples / Pitfalls

2.7.1 Example #1

Listing 11: example_01.ads

```

1 package Example_01 is
2
3     type Day is (Monday, Tuesday, Wednesday,
4                 Thursday, Friday, Saturday,
5                 Sunday);
6
7     subtype Weekend is Day range Saturday .. Sunday;
8
9     subtype Day_Off is Day range Wednesday | Weekend;
10
11 end Example_01;
```

This code is not correct. The syntax of range constraints does not allow sets of values. A predicate should be used instead.

2.7.2 Example #2

Listing 12: example_02.ads

```
1 package Example_02 is
2
3   type Day is (Monday, Tuesday, Wednesday,
4               Thursday, Friday, Saturday,
5               Sunday);
6
7   subtype Weekend is Day range Saturday .. Sunday;
8
9   subtype Day_Off is Weekend with
10      Static_Predicate => Day_Off in Wednesday | Weekend;
11
12 end Example_02;
```

This code is not correct. This is accepted by GNAT, but result is not the one expected by the user. Day_Off has the same constraint as Weekend.

2.7.3 Example #3

Listing 13: example_03.ads

```
1 package Example_03 is
2
3   type Day is (Monday, Tuesday, Wednesday,
4               Thursday, Friday, Saturday,
5               Sunday);
6
7   subtype Weekend is Day range Saturday .. Sunday;
8
9   subtype Day_Off is Day with
10      Dynamic_Predicate => Day_Off in Wednesday | Weekend;
11
12 end Example_03;
```

This code is correct. It is valid to use a Dynamic_Predicate where a Static_Predicate would be allowed.

2.7.4 Example #4

Listing 14: week.ads

```
1 package Week is
2
3   type Day is (Monday, Tuesday, Wednesday,
4               Thursday, Friday, Saturday,
5               Sunday);
6
7   subtype Weekend is Day range Saturday .. Sunday;
8
9   subtype Day_Off is Day with
```

(continues on next page)

(continued from previous page)

```

10     Static_Predicate => Day_Off in Wednesday | Weekend;
11
12 end Week;

```

Listing 15: example_04.adb

```

1  with Week; use Week;
2
3  procedure Example_04 is
4
5      function Next_Day_Off (D : Day_Off) return Day_Off is
6      begin
7          case D is
8              when Wednesday => return Saturday;
9              when Saturday  => return Sunday;
10             when Sunday    => return Wednesday;
11         end case;
12     end Next_Day_Off;
13
14 begin
15     null;
16 end Example_04;

```

This code is correct. It is valid to use a type with `Static_Predicate` for the value tested in a case statement. This is not true for `Dynamic_Predicate`.

2.7.5 Example #5

Listing 16: example_05.ads

```

1  package Example_05 is
2
3      type Day is (Monday, Tuesday, Wednesday,
4                  Thursday, Friday, Saturday,
5                  Sunday);
6
7      type Week_Schedule is private with
8          Type_Invariant => Valid (Week_Schedule);
9
10     function Valid (WS : Week_Schedule) return Boolean;
11
12 private
13     type Week_Schedule is record
14         Day_Off, Day_On_Duty : Day;
15     end record;
16
17     function Valid (WS : Week_Schedule) return Boolean is
18         (WS.Day_Off /= WS.Day_On_Duty);
19
20 end Example_05;

```

This code is correct. It is valid in Ada because the type invariant is not checked on entry or return from `Valid`. Also, function `Valid` is visible from the type invariant (special visibility in contracts). But it is invalid in SPARK, where private declaration cannot hold a type invariant. The reason is that the type invariant is assumed in the precondition of public functions for proof. That would lead to circular reasoning if `Valid` could be public.

2.7.6 Example #6

Listing 17: example_06.ads

```

1 package Example_06 is
2
3   type Day is (Monday, Tuesday, Wednesday,
4               Thursday, Friday, Saturday,
5               Sunday);
6
7   type Week_Schedule is private;
8
9 private
10
11  type Week_Schedule is record
12    Day_Off, Day_On_Duty : Day;
13  end record with
14    Type_Invariant => Valid (Week_Schedule);
15
16  function Valid (WS : Week_Schedule) return Boolean is
17    (WS.Day_Off /= WS.Day_On_Duty);
18
19 end Example_06;
```

This code is correct. This version is valid in both Ada and SPARK.

2.7.7 Example #7

Listing 18: example_07.ads

```

1 package Example_07 is
2
3   subtype Sorted_String is String with
4     Dynamic_Predicate =>
5       (for all Pos in Sorted_String'Range =>
6         Sorted_String (Pos) <= Sorted_String (Pos + 1));
7
8   subtype Unique_String is String with
9     Dynamic_Predicate =>
10      (for all Pos1, Pos2 in Unique_String'Range =>
11        Unique_String (Pos1) /= Unique_String (Pos2));
12
13  subtype Unique_Sorted_String is String with
14    Dynamic_Predicate =>
15      Unique_Sorted_String in Sorted_String and then
16      Unique_Sorted_String in Unique_String;
17
18 end Example_07;
```

This code is not correct. There are 3 problems in this code:

- there is a run-time error on the array access in Sorted_String;
- quantified expression defines only one variable;
- the property in Unique_String is true only for the empty string.

2.7.8 Example #8

Listing 19: example_08.ads

```

1 package Example_08 is
2
3   subtype Sorted_String is String with
4     Dynamic_Predicate =>
5       (for all Pos in Sorted_String'First ..
6         Sorted_String'Last - 1 =>
7           Sorted_String (Pos) <= Sorted_String (Pos + 1));
8
9   subtype Unique_String is String with
10    Dynamic_Predicate =>
11      (for all Pos1 in Unique_String'Range =>
12        (for all Pos2 in Unique_String'Range =>
13          (if Pos1 /= Pos2 then
14            Unique_String (Pos1) /= Unique_String (Pos2)))));
15
16   subtype Unique_Sorted_String is String with
17     Dynamic_Predicate =>
18       Unique_Sorted_String in Sorted_String and then
19       Unique_Sorted_String in Unique_String;
20
21 end Example_08;

```

This code is correct. This is a correct version in Ada. For proving AoRTE in SPARK, one will need to change slightly the property of Sorted_String.

2.7.9 Example #9

Listing 20: example_09.ads

```

1 package Example_09 is
2
3   type Day is (Monday, Tuesday, Wednesday,
4               Thursday, Friday, Saturday,
5               Sunday);
6
7   type Week_Schedule is private with
8     Default_Initial_Condition => Valid (Week_Schedule);
9
10  function Valid (WS : Week_Schedule) return Boolean;
11
12 private
13
14  type Week_Schedule is record
15    Day_Off, Day_On_Duty : Day;
16  end record;
17
18  function Valid (WS : Week_Schedule) return Boolean is
19    (WS.Day_Off /= WS.Day_On_Duty);
20
21 end Example_09;

```

This code is not correct. The default initial condition is not satisfied.

2.7.10 Example #10

Listing 21: example_10.ads

```
1 package Example_10 is
2
3   type Day is (Monday, Tuesday, Wednesday,
4               Thursday, Friday, Saturday,
5               Sunday);
6
7   type Week_Schedule is private with
8     Default_Initial_Condition => Valid (Week_Schedule);
9
10  function Valid (WS : Week_Schedule) return Boolean;
11
12 private
13
14  type Week_Schedule is record
15    Day_Off      : Day := Wednesday;
16    Day_On_Duty : Day := Friday;
17  end record;
18
19  function Valid (WS : Week_Schedule) return Boolean is
20    (WS.Day_Off /= WS.Day_On_Duty);
21
22 end Example_10;
```

This code is correct. This is a correct version, which can be proved with SPARK.

SYSTEMS PROGRAMMING

3.1 Type Contracts in Ada 2012 and SPARK 2014

3.2 Systems Programming - What is it?

- Bare metal programming
 - bare board applications (no Operating System)
 - Operating Systems (ex: Muen separation kernel)
 - device drivers (ex: Ada Drivers Library)
 - communication stacks (ex: AdaCore TCP/IP stack)
- Specifics of Systems Programming
 - direct access to hardware: registers, memory, etc.
 - side-effects (yes!)
 - efficiency is paramount (sometimes real-time even)
 - hard/impossible to debug

3.3 Systems Programming - How can SPARK help?

- SPARK is a Systems Programming language
 - same features as Ada for accessing hardware (representation clauses, address clauses)
 - as efficient as Ada or C
- Side-effects can be modeled in SPARK
 - reads and writes to memory-mapped devices are modeled
 - concurrent interactions with environment are modeled
- SPARK can help catch problems by static analysis
 - correct flows, initialization, concurrent accesses
 - absence of run-time errors and preservation of invariants

3.4 Systems Programming - A trivial example

Listing 1: show_trivial_sys_prog.ads

```

1 package Show_Trivial_Sys_Prog is
2
3   Y : Integer;
4
5   -- Y'Address could be replaced by any
6   -- external address
7   X : Integer with Volatile,
8     Address => Y'Address;
9
10  procedure Get (Val : out Integer)
11    with Global => (In_Out => X),
12     Depends => (Val => X,
13                X  => X);
14
15 end Show_Trivial_Sys_Prog;
```

Listing 2: show_trivial_sys_prog.adb

```

1 package body Show_Trivial_Sys_Prog is
2
3   procedure Get (Val : out Integer) is
4     begin
5       Val := X;
6     end Get;
7
8 end Show_Trivial_Sys_Prog;
```

- Comments:
 - X is volatile
 - X is also an output; output X depends on input X
 - X is only read

3.5 Volatile Variables and Volatile Types

- Variables whose reads/writes cannot be optimized away
- Identified through multiple aspects (or pragmas)
 - aspect `Volatile`
 - but also aspect `Atomic`
 - and GNAT aspect `Volatile_Full_Access`
 - all the above aspects can be set on type or object
- Other aspects are useful on volatile variables
 - aspect `Address` to specify location in memory
 - aspect `Import` to skip definition/initialization

```

type T is new Integer with Volatile;

X : Integer with Atomic, Import, Address => ... ;
```

3.6 Flavors of Volatile Variables

3.6.1 Using Async_Readers / Async_Writers

- Boolean aspects describing asynchronous behavior
 - Async_Readers if variable may be read asynchronously
 - Async_Writers if variable may be written asynchronously
- Effect of Async_Readers on flow analysis
- Effect of Async_Writers on flow analysis & proof
 - always initialized, always has an unknown value

Listing 3: volatile_vars.ads

```

1 package Volatile_Vars is
2
3   pragma Elaborate_Body;
4
5   Ext : array (1 .. 2) of Integer;
6
7   X : Integer with Volatile,
8       Address => Ext (1)'Address,
9       Async_Readers;
10
11  Y : Integer with Volatile,
12       Address => Ext (2)'Address,
13       Async_Writers;
14
15  procedure Set;
16 end Volatile_Vars;
```

Listing 4: volatile_vars.adb

```

1 package body Volatile_Vars is
2
3   procedure Set is
4     U, V : constant Integer := Y;
5   begin
6     pragma Assert (U = V);
7     X := 0;
8     X := 1;
9   end Set;
10 begin
11   Ext := (others => 0);
12 end Volatile_Vars;
```

Listing 5: show_volatile_vars.adb

```

1 with Volatile_Vars;
2
3 procedure Show_Volatile_Vars is
4 begin
5     Volatile_Vars.Set;
6 end Show_Volatile_Vars;

```

3.6.2 Using Effective_Reads / Effective_Writes

- Boolean aspects distinguishing values & sequences
 - Effective_Reads if reading the variable has an effect on its value
 - Effective_Writes if writing the variable has an effect on its value
- Effect of both on proof and flow dependencies
 - Final value of variable is seen as a sequence of values it took

Listing 6: volatile_vars.ads

```

1 package Volatile_Vars is
2
3     pragma Elaborate_Body;
4
5     Ext : array (1 .. 2) of Integer;
6
7     X : Integer with Volatile,
8         Address => Ext (1)'Address,
9         Async_Readers,
10        Effective_Writes;
11
12    Y : Integer with Volatile,
13        Address => Ext (2)'Address,
14        Async_Writers,
15        Effective_Reads;
16
17    procedure Set with
18        Depends => (X => Y,
19                    Y => Y);
20 end Volatile_Vars;

```

Listing 7: volatile_vars.adb

```

1 package body Volatile_Vars is
2
3     procedure Set is
4     begin
5         X := Y;
6         X := 0;
7     end Set;
8
9     begin
10        Ext := (others => 0);
11 end Volatile_Vars;

```

Listing 8: show_volatile_vars.adb

```

1 with Volatile_Vars;
2
3 procedure Show_Volatile_Vars is
4 begin
5   Volatile_Vars.Set;
6 end Show_Volatile_Vars;

```

3.6.3 Combinations of Flavors of Volatile Variables

- All four flavors can be set independently
 - Default for Volatile/Atomic is all four **True**
 - When some aspects set, all others default to **False**
- Only half the possible combinations are legal
 - Async_Readers and/or Async_Writers is set
 - Effective_Reads = **True** forces Async_Writers = **True**
 - Effective_Writes = **True** forces Async_Readers = **True**
 - sensor: AW=**True**
 - actuator: AR=**True**
 - input port: AW=**True**, ER=**True**
 - output port: AR=**True**, EW=**True**

3.7 Constraints on Volatile Variables

- Volatile variables must be defined at library level
- Expressions (and functions) cannot have side-effects
 - read of variable with AW=**True** must appear alone on *rhs* of assign
 - a function cannot read a variable with ER=**True**

Listing 9: volatile_vars.ads

```

1 package Volatile_Vars is
2
3   pragma Elaborate_Body;
4
5   Ext : array (1 .. 4) of Integer;
6
7   AR : Integer with Volatile,
8       Address => Ext (1)'Address,
9       Async_Readers;
10
11  AW : Integer with Volatile,
12      Address => Ext (2)'Address,
13      Async_Writers;
14
15  ER : Integer with Volatile,
16      Address => Ext (3)'Address,

```

(continues on next page)

(continued from previous page)

```

17     Async_Writers,
18     Effective_Reads;
19
20     EW : Integer with Volatile,
21     Address => Ext (4)'Address,
22     Async_Readers,
23     Effective_Writes;
24
25     procedure Read_All;
26
27     function Read_ER return Integer;
28
29     procedure Set (V : Integer);
30
31 end Volatile_Vars;

```

Listing 10: volatile_vars.adb

```

1  package body Volatile_Vars is
2
3     procedure Read_All is
4         Tmp : Integer := 0;
5     begin
6         Tmp := Tmp + AR;
7         Tmp := Tmp + AW;
8         EW := Tmp;
9         Set (ER);
10    end Read_All;
11
12    function Read_ER return Integer is
13        Tmp : Integer := ER;
14    begin
15        return Tmp;
16    end Read_ER;
17
18    procedure Set (V : Integer) is
19    begin
20        AW := V;
21    end Set;
22
23    begin
24        Ext := (others => 0);
25    end Volatile_Vars;

```

Listing 11: show_volatile_vars.adb

```

1  with Volatile_Vars;
2
3  procedure Show_Volatile_Vars is
4      V : Integer;
5  begin
6      Volatile_Vars.Read_All;
7      V := Volatile_Vars.Read_ER;
8  end Show_Volatile_Vars;

```

- Comments:
 - AW not alone on rhs
 - ER not alone on rhs
 - ER output of Read_ER

3.8 Constraints on Volatile Functions

- Functions should have mathematical interpretation
 - a function reading a variable with AW=**True** is marked as volatile with aspect `Volatile_Function`
 - calls to volatile functions are restricted like reads of `Async_Writers`

Listing 12: volatile_vars.ads

```

1 package Volatile_Vars is
2
3   pragma Elaborate_Body;
4
5   Ext : array (1 .. 4) of Integer;
6
7   AR : Integer with Volatile,
8       Address => Ext (1)'Address,
9       Async_Readers;
10
11  AW : Integer with Volatile,
12      Address => Ext (2)'Address,
13      Async_Writers;
14
15  ER : Integer with Volatile,
16      Address => Ext (3)'Address,
17      Async_Writers,
18      Effective_Reads;
19
20  EW : Integer with Volatile,
21      Address => Ext (4)'Address,
22      Async_Readers,
23      Effective_Writes;
24
25  function Read_Non_Volatile
26    return Integer;
27
28  function Read_Volatile
29    return Integer
30    with Volatile_Function;
31
32  function Read_ER
33    return Integer
34    with Volatile_Function;
35
36 end Volatile_Vars;
```

Listing 13: volatile_vars.adb

```

1 package body Volatile_Vars is
2
3   function Read_Non_Volatile
4     return Integer is
5     Tmp : Integer := 0;
6   begin
7     -- reads AR, AW, EW
8     -- ERROR: not a volatile function
9     Tmp := Tmp + AR;
10    Tmp := Tmp + AW;
11    Tmp := Tmp + EW;
12
```

(continues on next page)

(continued from previous page)

```

13     return Tmp;
14 end Read_Non_Volatile;
15
16 function Read_Volatile
17   return Integer is
18     Tmp : Integer := 0;
19   begin
20     -- reads AR, AW, EW
21     -- OK for volatile function
22     Tmp := Tmp + AR;
23     Tmp := Tmp + AW;
24     Tmp := Tmp + EW;
25
26     return Tmp;
27 end Read_Volatile;
28
29 function Read_ER
30   return Integer is
31     Tmp : Integer := ER;
32   begin
33     -- reads ER
34     -- ERROR: ER output of Read_ER
35     return Tmp;
36 end Read_ER;
37
38 begin
39   Ext := (others => 0);
40 end Volatile_Vars;

```

Listing 14: show_volatile_vars.adb

```

1 with Volatile_Vars;
2
3 procedure Show_Volatile_Vars is
4   V : Integer;
5   begin
6     V := Volatile_Vars.Read_Non_Volatile;
7     V := Volatile_Vars.Read_Volatile;
8     V := Volatile_Vars.Read_ER;
9 end Show_Volatile_Vars;

```

3.9 State Abstraction on Volatile Variables

- Abstract state needs to be identified as External
- Flavors of volatility can be specified
 - Default if none specified is all True

Listing 15: p1.ads

```

1 package P1 with
2   Abstract_State => (S with External)
3 is
4   procedure Process (Data : out Integer) with
5     Global => (In_Out => S);
6
7 end P1;

```

Listing 16: p2.ads

```

1 package P2 with
2   Abstract_State => (S with External =>
3     (Async_Writers,
4       -- OK if refined into AW, ER
5       Effective_Reads)
6     -- not OK if refined into AR, EW
7     )
8 is
9   procedure Process (Data : out Integer) with
10     Global => (In_Out => S);
11
12 end P2;
```

3.10 Constraints on Address Attribute

- Address of volatile variable can be specified

Listing 17: show_address_attribute.ads

```

1 package Show_Address_Attribute is
2
3   Ext : array (1 .. 2) of Integer;
4
5   X : Integer with Volatile,
6     Address => Ext (1)'Address;
7
8   Y : Integer with Volatile;
9   for Y'Address use Ext (2)'Address;
10
11 end Show_Address_Attribute;
```

- Address attribute not allowed in expressions
- Overlays are allowed
 - GNATprove does not check absence of overlays
 - GNATprove does not model the resulting aliasing

Listing 18: show_address_overlay.adb

```
1 procedure Show_Address_Overlay is
2
3   X : Integer := 1;
4   Y : Integer := 0
5     with Address => X'Address;
6
7   pragma Assert (X = 1);
8   -- assertion wrongly proved
9 begin
10  null;
11
12 end Show_Address_Overlay;
```

3.11 Can something be known of volatile variables?

- Variables with Async_Writers have no known value
- ... but they have a known type!
 - type range, ex: 0 .. 360
 - type predicate, ex: 0 .. 15 | 17 .. 42 | 43 .. 360
- Variables without Async_Writers have a known value
- GNATprove also assumes all values are valid (X'Valid)

Listing 19: show_provable_volatile_var.ads

```
1 package Show_Provable_Volatile_Var is
2
3   X : Integer with Volatile, Async_Readers;
4
5   procedure Read_Value;
6
7 end Show_Provable_Volatile_Var;
```

Listing 20: show_provable_volatile_var.adb

```

1 package body Show_Provable_Volatile_Var is
2
3   procedure Read_Value is
4   begin
5     X := 42;
6     pragma Assert (X = 42);
7     -- proved!
8   end Read_Value;
9
10 end Show_Provable_Volatile_Var;
```

3.12 Other Concerns in Systems Programming

- Software startup state → elaboration rules
 - SPARK follows Ada static elaboration model
 - ... with additional constraints for ensuring correct initialization
 - ... but GNATprove follows the relaxed GNAT static elaboration
- Handling of faults → exception handling
 - raising exceptions is allowed in SPARK
 - ... but exception handlers are SPARK_Mode => Off
 - ... typically the last-chance-handler is used instead
- Concurrency inside the application → tasking support
 - Ravenscar and Extended_Ravenscar profiles supported in SPARK

3.13 Code Examples / Pitfalls

3.13.1 Example #1

Listing 21: example_01.ads

```

1 package Example_01 is
2
3   Ext : Integer;
4
5   X   : Integer with Volatile,
6       Address => Ext'Address;
7
8   procedure Get (Val : out Integer)
9     with Global => (Input => X),
10    Depends => (Val => X);
11
12 end Example_01;
```

Listing 22: example_01.adb

```
1 package body Example_01 is
2
3   procedure Get (Val : out Integer) is
4     begin
5       Val := X;
6     end Get;
7
8 end Example_01;
```

This code is not correct. X has Effective_Reads set by default, hence it is also an output.

3.13.2 Example #2

Listing 23: example_02.ads

```
1 package Example_02 is
2
3   Ext : Integer;
4
5   X : Integer with Volatile, Address => Ext'Address,
6     Async_Readers, Async_Writers, Effective_Writes;
7
8   procedure Get (Val : out Integer)
9     with Global => (Input => X),
10    Depends => (Val => X);
11
12 end Example_02;
```

Listing 24: example_02.adb

```
1 package body Example_02 is
2
3   procedure Get (Val : out Integer) is
4     begin
5       Val := X;
6     end Get;
7
8 end Example_02;
```

This code is correct. X has Effective_Reads = **False**, hence it is only an input.

3.13.3 Example #3

Listing 25: example_03.ads

```
1 package Example_03 is
2
3   Speed : Float with Volatile, Async_Writers;
4   Motor : Float with Volatile, Async_Readers;
5
6   procedure Adjust with
7     Depends => (Motor =>+ Speed);
8
9 end Example_03;
```

Listing 26: example_03.adb

```

1 package body Example_03 is
2
3   procedure Adjust is
4     Cur_Speed : constant Float := Speed;
5   begin
6     if abs (Cur_Speed) > 100.0 then
7       Motor := Motor - 1.0;
8     end if;
9   end Adjust;
10
11 end Example_03;

```

This code is correct. Speed is an input only, Motor is both an input and output. Note how the current value of Speed is first copied to be tested in a larger expression.

3.13.4 Example #4

Listing 27: example_04.ads

```

1 package Example_04 is
2
3   Raw_Data : Float with Volatile,
4     Async_Writers, Effective_Reads;
5   Data      : Float with Volatile,
6     Async_Readers, Effective_Writes;
7
8   procedure Smooth with
9     Depends => (Data => Raw_Data);
10
11 end Example_04;

```

Listing 28: example_04.adb

```

1 package body Example_04 is
2
3   procedure Smooth is
4     Data1 : constant Float := Raw_Data;
5     Data2 : constant Float := Raw_Data;
6   begin
7     Data := Data1;
8     Data := (Data1 + Data2) / 2.0;
9     Data := Data2;
10  end Smooth;
11
12 end Example_04;

```

This code is not correct. Raw_Data has Effective_Reads set, hence it is also an output.

3.13.5 Example #5

Listing 29: example_05.ads

```
1 package Example_05 is
2
3   type Regval is new Integer with Volatile;
4   type Regnum is range 1 .. 32;
5   type Registers is array (Regnum) of Regval;
6
7   Regs : Registers with Async_Writers, Async_Readers;
8
9   function Reg (R : Regnum) return Integer is
10     (Integer (Regs (R))) with Volatile_Function;
11
12 end Example_05;
```

This code is not correct. Regs has Async_Writers set, hence it cannot appear as the expression in an expression function.

3.13.6 Example #6

Listing 30: example_06.ads

```
1 package Example_06 is
2
3   type Regval is new Integer with Volatile;
4   type Regnum is range 1 .. 32;
5   type Registers is array (Regnum) of Regval;
6
7   Regs : Registers with Async_Writers, Async_Readers;
8
9   function Reg (R : Regnum) return Integer
10     with Volatile_Function;
11
12 end Example_06;
```

Listing 31: example_06.adb

```

1 package body Example_06 is
2
3   function Reg (R : Regnum) return Integer is
4     V : Regval := Regs (R);
5   begin
6     return Integer (V);
7   end Reg;
8
9 end Example_06;

```

This code is not correct. `Regval` is a volatile type, hence variable `V` is volatile and cannot be declared locally.

3.13.7 Example #7

Listing 32: example_07.ads

```

1 package Example_07 is
2
3   type Regval is new Integer with Volatile;
4   type Regnum is range 1 .. 32;
5   type Registers is array (Regnum) of Regval;
6
7   Regs : Registers with Async_Writers, Async_Readers;
8
9   function Reg (R : Regnum) return Integer
10    with Volatile_Function;
11
12 end Example_07;

```

Listing 33: example_07.adb

```

1 package body Example_07 is
2
3   function Reg (R : Regnum) return Integer is
4   begin
5     return Integer (Regs (R));
6   end Reg;
7
8 end Example_07;

```

This code is correct. `Regs` has `Effective_Reads = False` hence can be read in a function. Function `Reg` is marked as volatile with aspect `Volatile_Function`. No volatile variable is declared locally.

3.13.8 Example #8

Listing 34: example_08.ads

```

1 package Example_08 with
2   Abstract_State => (State with External),
3   Initializes => State
4 is
5   procedure Dummy;
6 end Example_08;

```

Listing 35: example_08.adb

```

1 package body Example_08 with
2   Refined_State => (State => (X, Y, Z))
3 is
4   X : Integer with Volatile, Async_Readers;
5   Y : Integer with Volatile, Async_Writers;
6   Z : Integer := 0;
7
8   procedure Dummy is
9   begin
10    null;
11  end Dummy;
12
13 end Example_08;

```

This code is not correct. X has `Async_Writers = False`, hence is not considered as always initialized. As aspect `Initializes` specifies that `State` should be initialized after elaboration, this is an error. Note that it is allowed to bundle volatile and non-volatile variables in an external abstract state.

3.13.9 Example #9

Listing 36: example_09.ads

```

1 package Example_09 is
2
3   type Pair is record
4     U, V : Natural;
5   end record
6   with Predicate => U /= V;
7
8   X : Pair with Atomic, Async_Readers, Async_Writers;
9
10  function Max return Integer with
11    Volatile_Function,
12    Post => Max'Result /= 0;
13
14 end Example_09;

```

Listing 37: example_09.adb

```

1 package body Example_09 is
2
3   function Max return Integer is
4     Val1 : constant Natural := X.U;
5     Val2 : constant Natural := X.V;
6   begin

```

(continues on next page)

(continued from previous page)

```

7     return Natural'Max (Val1, Val2);
8   end Max;
9
10  end Example_09;

```

This code is not correct. X has Async_Writers set, hence it may have been written between the successive reads of X.U and X.V.

3.13.10 Example #10

Listing 38: example_10.ads

```

1  package Example_10 is
2
3     type Pair is record
4       U, V : Natural;
5     end record
6     with Predicate => U /= V;
7
8     X : Pair with Atomic, Async_Readers, Async_Writers;
9
10    function Max return Integer with
11      Volatile_Function,
12      Post => Max'Result /= 0;
13
14  end Example_10;

```

Listing 39: example_10.adb

```

1  package body Example_10 is
2
3     function Max return Integer is
4       P      : constant Pair := X;
5       Val1   : constant Natural := P.U;
6       Val2   : constant Natural := P.V;
7     begin
8       return Natural'Max (Val1, Val2);
9     end Max;
10
11  end Example_10;

```


This code is correct. Values of P.U and P.V are provably different, and the postcondition is proved.

CONCURRENCY

4.1 Concurrency \neq Parallelism

- Concurrency allows to create a well structured program
- Parallelism allows to create a high performance program
- Multiple cores/processors are...
 - possible for concurrent programs
 - essential to parallelism
- What about Ada and SPARK?
 - GNAT runtimes for concurrency available on single core & multicore (for SMP platforms)
 - parallel features scheduled for inclusion in Ada and SPARK 202x

4.2 Concurrent Program Structure in Ada




4.3 The problems with concurrency

- Control and data flow become much more complex
 - possibly nondeterministic even
 - actual behavior is one of many possible interleavings of tasks
- Data may be corrupted by concurrent accesses
 - so called data races or race conditions
- Control may block indefinitely, or loop indefinitely
 - so called deadlocks and livelocks
- Scheduling and memory usage are harder to compute

4.4 Ravenscar - the Ada solution to concurrency problems

- Ravenscar profile restricts concurrency in Ada
 - ensures deterministic behavior at every point in time
 - recommends use of protected objects to avoid data races
 - prevents deadlocks with Priority Ceiling Protocol
 - allows use of scheduling analysis techniques (RMA, RTA)
 - facilitates computation of memory usage with static tasks
- GNAT Extended Ravenscar profile lifts some restrictions
 - still same benefits as Ravenscar profile
 - removes painful restrictions for some applications

4.5 Concurrent Program Structure in Ravenscar



4.6 Ravenscar - the SPARK solution to concurrency problems

- Ravenscar and Extended_Ravenscar profiles supported in SPARK
- Data races prevented by flow analysis
 - ensures no problematic concurrent access to unprotected data
 - flow analysis also ensures non-termination of tasks
- Run-time errors prevented by proof
 - includes violations of the Priority Ceiling Protocol

4.7 Concurrency - A trivial example

Listing 1: show_trivial_task.ads

```

1 package Show_Trivial_Task is
2   type Task_Id is new Integer;
3   task type T (Id : Task_Id);
4
5   T1 : T (0);
6   T2 : T (1);
7 end Show_Trivial_Task;
```

Listing 2: show_trivial_task.adb

```
1 package body Show_Trivial_Task is
2   task body T is
3     Current_Task : Task_Id := Id;
4     begin
5       loop
6         delay 1.0;
7       end loop;
8     end T;
9 end Show_Trivial_Task;
```

- Id can be written by T1 and T2 at the same time

4.8 Setup for using concurrency in SPARK

- Any unit using concurrency features (tasks, protected objects, etc.) must set the profile

```
pragma Profile (Ravenscar);
-- or
pragma Profile (GNAT_Extended_Ravenscar);
```

- ... plus an additional pragma
 - that ensures tasks start after the end of elaboration

```
pragma Partition_Elaboration_Policy (Sequential);
```

- ... which are checked by GNAT partition-wide
 - pragmas needed for verification even if not for compilation

4.9 Tasks in Ravenscar

- A task can be either a singleton object or a type
 - no declarations of entries for rendez-vous

```
task T;
task type TT;
```

- ... completed by a body
 - infinite loop to prevent termination

```
task body T is
begin
  loop
    ...
  end loop;
end T;
```

- Tasks are declared at library-level
- ... as standalone objects or inside records/arrays

```

type TA is array (1 .. 3) of TT;
type TR is record
  A, B : TT;
end record;

```

4.10 Communication Between Tasks in Ravenscar

- Tasks can communicate through protected objects
- A protected object is either a singleton object or a type
 - all PO private data initialized by default in SPARK

Listing 3: show_protected_object.ads

```

1 package Show_Protected_Object is
2
3   protected P is
4     procedure Set (V : Natural);
5     function Get return Natural;
6   private
7     The_Data : Natural := 0;
8   end P;
9
10 end Show_Protected_Object;

```

Listing 4: show_protected_object.adb

```

1 package body Show_Protected_Object is
2
3   protected body P is
4     procedure Set (V : Natural) is
5       begin
6         The_Data := V;
7       end Set;
8     function Get return Natural is
9       (The_Data);
10    end P;
11
12 end Show_Protected_Object;

```

4.11 Protected Objects in Ravenscar

- Protected objects are declared at library-level
- ... as standalone objects or inside records/arrays
 - The record type needs to be volatile, as a non-volatile type cannot contain a volatile component. The array type is implicitly volatile when its component type is volatile.

Listing 5: show_protected_object_ravenscar.ads

```

1 package Show_Protected_Object_Ravenscar is
2
3   protected type PT is

```

(continues on next page)

(continued from previous page)

```

4     procedure Set (V : Natural);
5     function Get return Natural;
6 private
7     The_Data : Natural := 0;
8 end PT;
9
10    P : PT;
11
12    type PAT is array (1 .. 3) of PT;
13    PA : PAT;
14
15    type PRT is record
16        A, B : PT;
17    end record with Volatile;
18    PR : PRT;
19
20 end Show_Protected_Object_Ravenscar;

```

Listing 6: show_protected_object_ravenscar.adb

```

1 package body Show_Protected_Object_Ravenscar is
2
3     protected body PT is
4         procedure Set (V : Natural) is
5             begin
6                 The_Data := V;
7             end Set;
8         function Get return Natural is
9             (The_Data);
10        end PT;
11
12 end Show_Protected_Object_Ravenscar;

```

4.12 Protected Communication with Procedures & Functions

- CREW enforced (Concurrent-Read-Exclusive-Write)
 - procedures have exclusive read-write access to PO
 - functions have shared read-only access to PO
- Actual mechanism depends on target platform
 - scheduler enforces policy on single core
 - locks used on multicore (using CAS instructions)
 - lock-free transactions used for simple PO (again using CAS)
- Mechanism is transparent to user
 - user code simply calls procedures/functions
 - task may be queued until PO is released by another task

4.13 Blocking Communication with Entries

- Only protected objects have entries in Ravenscar
- Entry = procedure with **entry** guard condition
 - second level of queues, one for each entry, on a given PO
 - task may be queued until guard is True and PO is released
 - at most one entry in Ravenscar
 - guard is a **Boolean** component of PO in Ravenscar

Listing 7: show_blocking_communication.ads

```

1 package Show_Blocking_Communication is
2
3   protected type PT is
4     entry Reset;
5   private
6     Is_Not_Null : Boolean := False;
7     The_Data    : Integer := 1000;
8   end PT;
9
10 end Show_Blocking_Communication;
```

Listing 8: show_blocking_communication.adb

```

1 package body Show_Blocking_Communication is
2
3   protected body PT is
4     entry Reset when Is_Not_Null is
5     begin
6       The_Data := 0;
7     end Reset;
8   end PT;
9
10 end Show_Blocking_Communication;
```

4.14 Relaxed Constraints on Entries with Extended Ravenscar

- Proof limitations with Ravenscar
 - not possible to relate guard to other components with invariant
- GNAT Extended Ravenscar profile lifts these constraints
 - and allows multiple tasks to call the same entry

Listing 9: show_relaxed_constraints_on_entries.ads

```

1 package Show_Relaxed_Constraints_On_Entries is
2
3   protected type Mailbox is
4     entry Publish;
5     entry Retrieve;
6   private
7     Num_Messages : Natural := 0;
```

(continues on next page)

(continued from previous page)

```

8   end Mailbox;
9
10  end Show_Relaxed_Constraints_On_Entries;

```

Listing 10: show_relaxed_constraints_on_entries.adb

```

1  package body Show_Relaxed_Constraints_On_Entries is
2
3     Max : constant := 100;
4
5     protected body Mailbox is
6         entry Publish when Num_Messages < Max is
7             begin
8                 Num_Messages := Num_Messages + 1;
9             end Publish;
10
11        entry Retrieve when Num_Messages > 0 is
12            begin
13                Num_Messages := Num_Messages - 1;
14            end Retrieve;
15        end Mailbox;
16
17  end Show_Relaxed_Constraints_On_Entries;

```

4.15 Interrupt Handlers in Ravenscar

- Interrupt handlers are parameterless procedures of PO
 - with aspect `Attach_Handler` specifying the corresponding signal
 - with aspect `Interrupt_Priority` on the PO specifying the priority

Listing 11: show_interrupt_handlers.ads

```

1  with System; use System;
2  with Ada.Interrupts.Names; use Ada.Interrupts.Names;
3
4  package Show_Interrupt_Handlers is
5
6     protected P with
7         Interrupt_Priority =>
8             System.Interrupt_Priority'First
9     is
10        procedure Signal with
11            Attach_Handler => SIGHUP;
12        end P;
13
14  end Show_Interrupt_Handlers;

```

- Priority of the PO should be in `System.Interrupt_Priority`
 - default is OK - in the range of `System.Interrupt_Priority`
 - checked by proof (default or value of `Priority` or `Interrupt_Priority`)

4.16 Other Communications Between Tasks in SPARK

- Tasks must communicate through synchronized objects
 - atomic objects
 - protected objects
 - suspension objects (standard **Boolean** protected objects)
- Constants are considered as synchronized
 - this includes variables constant after elaboration (specified with aspect `Constant_After_Elaboration`)
- Single task or PO can access an unsynchronized object
 - exclusive relation between object and task/PO must be specified with aspect `Part_Of`

4.17 Data and Flow Dependencies of Tasks

- Input/output relation can be specified for a task
 - as task never terminates, output is understood while task runs
 - task itself is both an input and an output
 - implicit `In_Out => T`
 - explicit dependency

Listing 12: `show_data_and_flow_dependencies.ads`

```

1 package Show_Data_And_Flow_Dependencies is
2
3   X, Y, Z : Integer;
4
5   task T with
6     Global => (Input => X,
7               Output => Y,
8               In_Out => Z),
9     Depends => (T => T,
10              Z => X,
11              Y => X,
12              null => Z);
13 end Show_Data_And_Flow_Dependencies;
```

4.18 State Abstraction over Synchronized Variables

- Synchronized objects can be abstracted in synchronized abstract state with aspect `Synchronous`

Listing 13: `show_state_abstraction.ads`

```

1 package Show_State_Abstraction with
2   Abstract_State => (State with Synchronous, External)
3 is
4
```

(continues on next page)

(continued from previous page)

```

5   protected type Protected_Type is
6       procedure Reset;
7   private
8       Data : Natural := 0;
9   end Protected_Type;
10
11  task type Task_Type;
12
13  end Show_State_Abstraction;

```

Listing 14: show_state_abstraction.adb

```

1  package body Show_State_Abstraction with
2     Refined_State => (State => (A, P, T))
3  is
4     A : Integer with Atomic, Async_Readers, Async_Writers;
5     P : Protected_Type;
6     T : Task_Type;
7
8     protected body Protected_Type is
9         procedure Reset is
10            begin
11                Data := 0;
12            end Reset;
13     end Protected_Type;
14
15     task body Task_Type is
16     begin
17         P.Reset;
18         A := 0;
19     end Task_Type;
20
21  end Show_State_Abstraction;

```

- Synchronized state is a form of external state
 - Synchronous same as External => (Async_Readers, Async_Writers)
 - tasks are not volatile and can be part of regular abstract state

4.19 Synchronized Abstract State in the Standard Library

- Standard library maintains synchronized state
 - the tasking runtime maintains state about running tasks
 - the real-time runtime maintains state about current time

```

package Ada.Task_Identification with
  SPARK_Mode,
  Abstract_State =>
    (Tasking_State with Synchronous,
     External => (Async_Readers, Async_Writers)),
  Initializes => Tasking_State

package Ada.Real_Time with
  SPARK_Mode,
  Abstract_State =>

```

(continues on next page)

(continued from previous page)

```
(Clock_Time with Synchronous,
  External => (Async_Readers, Async_Writers)),
Initializes => Clock_Time
```

- API of these units refer to **Tasking_State** and **Clock_Time**

4.20 Code Examples / Pitfalls

4.20.1 Example #1

Listing 15: rendezvous.adb

```
1 procedure Rendezvous is
2   task T1 is
3     entry Start;
4   end T1;
5
6   task body T1 is
7   begin
8     accept Start;
9   end T1;
10
11 begin
12   T1.Start;
13 end Rendezvous;
```

This code is not correct. Task rendezvous is not allowed; violation of restriction `Max_Task_Entries = 0`. A local task is not allowed; violation of restriction `No_Task_Hierarchy`

4.20.2 Example #2

Listing 16: example_02.ads

```
1 package Example_02 is
2
3   protected P is
4     entry Reset;
5   end P;
6
7 private
8   Data : Boolean := False;
9 end Example_02;
```

Listing 17: example_02.adb

```
1 package body Example_02 is
2
3   protected body P is
4     entry Reset when Data is
5     begin
6       null;
7     end Reset;
8   end P;
```

(continues on next page)

(continued from previous page)

```
9  
10 end Example_02;
```

This code is not correct. Global data in entry guard is not allowed. Violation of restriction Simple_Barriers (for Ravenscar) or Pure_Barriers (for Extended Ravenscar)

4.20.3 Example #3

Listing 18: example_03.ads

```
1 package Example_03 is  
2  
3   protected P is  
4     procedure Set (Value : Integer);  
5   end P;  
6  
7   private  
8     task type TT;  
9  
10    T1, T2 : TT;  
11  
12 end Example_03;
```

Listing 19: example_03.adb

```
1 package body Example_03 is  
2  
3   Data : Integer := 0;  
4  
5   protected body P is  
6     procedure Set (Value : Integer) is  
7       begin  
8         Data := Value;  
9       end Set;  
10    end P;  
11  
12    task body TT is  
13      Local : Integer := 0;  
14      begin  
15        loop  
16          Local := (Local + 1) mod 100;  
17          P.Set (Local);  
18        end loop;  
19      end TT;  
20  
21 end Example_03;
```

This code is not correct. Global unprotected data accessed in protected object shared between tasks

4.20.4 Example #4

Listing 20: example_04.ads

```

1 package Example_04 is
2
3   protected P is
4     procedure Set (Value : Integer);
5   end P;
6
7 private
8   Data : Integer := 0 with Part_Of => P;
9
10  task type TT;
11
12  T1, T2 : TT;
13
14 end Example_04;
```

Listing 21: example_04.adb

```

1 package body Example_04 is
2
3   protected body P is
4     procedure Set (Value : Integer) is
5       begin
6         Data := Value;
7       end Set;
8   end P;
9
10  task body TT is
11    Local : Integer := 0;
12  begin
13    loop
14      Local := (Local + 1) mod 100;
15      P.Set (Local);
16    end loop;
17  end TT;
18
19 end Example_04;
```

This code is correct. Data is part of the protected object state. The only accesses to Data are through P.

4.20.5 Example #5

Listing 22: example_05.ads

```

1 package Example_05 is
2
3   protected P1 with Priority => 3 is
4     procedure Set (Value : Integer);
5   private
6     Data : Integer := 0;
7   end P1;
8
9   protected P2 with Priority => 2 is
10    procedure Set (Value : Integer);
11  end P2;
```

(continues on next page)

(continued from previous page)

```

12
13 private
14   task type TT with Priority => 1;
15
16   T1, T2 : TT;
17
18 end Example_05;

```

Listing 23: example_05.adb

```

1 package body Example_05 is
2
3   protected body P1 is
4     procedure Set (Value : Integer) is
5       begin
6         Data := Value;
7       end Set;
8   end P1;
9
10  protected body P2 is
11    procedure Set (Value : Integer) is
12      begin
13        P1.Set (Value);
14      end Set;
15    end P2;
16
17  task body TT is
18    Local : constant Integer := 0;
19  begin
20    loop
21      P2.Set (Local);
22    end loop;
23  end TT;
24
25 end Example_05;

```

This code is correct. Ceiling_Priority policy is respected. Task never accesses a protected object with lower priority than its active priority. Note that PO can call procedure or function from another PO, but not an entry (possibly blocking).

4.20.6 Example #6

Listing 24: example_06.ads

```

1 package Example_06 is
2
3   protected type Mailbox is
4     entry Publish;
5     entry Retrieve;
6   private
7     Not_Empty    : Boolean := True;
8     Not_Full     : Boolean := False;
9     Num_Messages : Natural := 0;
10  end Mailbox;
11
12 end Example_06;

```

Listing 25: example_06.adb

```

1 package body Example_06 is
2
3   Max : constant := 100;
4
5   protected body Mailbox is
6     entry Publish when Not_Full is
7       begin
8         Num_Messages := Num_Messages + 1;
9         Not_Empty := True;
10        if Num_Messages = Max then
11          Not_Full := False;
12        end if;
13      end Publish;
14
15     entry Retrieve when Not_Empty is
16       begin
17         Num_Messages := Num_Messages - 1;
18         Not_Full := True;
19         if Num_Messages = 0 then
20           Not_Empty := False;
21         end if;
22      end Retrieve;
23   end Mailbox;
24
25 end Example_06;

```

This code is not correct. Integer range cannot be proved correct.

4.20.7 Example #7

Listing 26: example_07.ads

```

1 package Example_07 is
2
3   protected type Mailbox is
4     entry Publish;
5     entry Retrieve;
6   private
7     Num_Messages : Natural := 0;
8   end Mailbox;
9
10 end Example_07;

```

Listing 27: example_07.adb

```

1 package body Example_07 is
2
3   Max : constant := 100;
4
5   protected body Mailbox is
6     entry Publish when Num_Messages < Max is
7       begin
8         Num_Messages := Num_Messages + 1;
9       end Publish;
10
11     entry Retrieve when Num_Messages > 0 is
12       begin

```

(continues on next page)

(continued from previous page)

```

13     Num_Messages := Num_Messages - 1;
14   end Retrieve;
15 end Mailbox;
16
17 end Example_07;

```

This code is correct. Precise range obtained from entry guards allows to prove checks.

4.20.8 Example #8

Listing 28: example_08.ads

```

1 package Example_08 is
2
3   Max : constant := 100;
4
5   type Content is record
6     Not_Empty   : Boolean := False;
7     Not_Full    : Boolean := True;
8     Num_Messages : Natural := 0;
9   end record with Predicate =>
10    Num_Messages in 0 .. Max
11    and Not_Empty = (Num_Messages > 0)
12    and Not_Full = (Num_Messages < Max);
13
14   protected type Mailbox is
15     entry Publish;
16     entry Retrieve;
17   private
18     C : Content;
19   end Mailbox;
20
21 end Example_08;

```

Listing 29: example_08.adb

```

1 package body Example_08 is
2
3   protected body Mailbox is
4     entry Publish when C.Not_Full is
5       Not_Full      : Boolean := C.Not_Full;
6       Num_Messages : Natural := C.Num_Messages;
7     begin
8       Num_Messages := Num_Messages + 1;
9       if Num_Messages = Max then
10        Not_Full := False;
11      end if;
12      C := (True, Not_Full, Num_Messages);
13    end Publish;
14
15     entry Retrieve when C.Not_Empty is
16       Not_Empty      : Boolean := C.Not_Empty;
17       Num_Messages : Natural := C.Num_Messages;
18     begin
19       Num_Messages := Num_Messages - 1;
20       if Num_Messages = 0 then
21        Not_Empty := False;
22      end if;
23      C := (Not_Empty, True, Num_Messages);

```

(continues on next page)

(continued from previous page)

```

24     end Retrieve;
25     end Mailbox;
26
27 end Example_08;

```

This code is correct. Precise range obtained from predicate allows to prove checks. Predicate is preserved.

4.20.9 Example #9

Listing 30: example_09.ads

```

1  --% src_file: Example_09.ads
2  --% cflags: -gnaty
3  --% make_flags: -gnaty -gnata
4
5  package Example_09 is
6
7      package Service with
8          Abstract_State => (State with External)
9      is
10         procedure Extract (Data : out Integer) with
11             Global => (In_Out => State);
12         end Service;
13
14     private
15         task type T;
16         T1, T2 : T;
17
18     end Example_09;

```

Listing 31: example_09.adb

```

1  package body Example_09 is
2
3      package body Service with
4          Refined_State => (State => Extracted)
5      is
6          Local_Data : constant Integer := 100;
7          Extracted : Boolean := False;
8
9          procedure Extract (Data : out Integer) is
10             begin
11                 if not Extracted then
12                     Data := Local_Data;
13                     Extracted := True;
14                 else
15                     Data := Integer'First;
16                 end if;
17             end Extract;
18         end Service;
19
20         task body T is
21             X : Integer;
22             begin
23                 loop
24                     Service.Extract (X);
25                 end loop;

```

(continues on next page)

(continued from previous page)

```

26   end T;
27
28 end Example_09;

```

This code is not correct. Unsynchronized state cannot be accessed from multiple tasks or protected objects.

4.20.10 Example #10

Listing 32: example_10.ads

```

1 package Example_10 is
2
3   package Service with
4     Abstract_State => (State with Synchronous, External)
5   is
6     procedure Extract (Data : out Integer) with
7       Global => (In_Out => State);
8   private
9     protected type Service_Extracted is
10      procedure Set;
11      function Get return Boolean;
12    private
13      Extracted : Boolean := False;
14    end Service_Extracted;
15  end Service;
16
17 private
18   task type T;
19   T1, T2 : T;
20
21 end Example_10;

```

Listing 33: example_10.adb

```

1 package body Example_10 is
2
3   package body Service with
4     Refined_State => (State => Extracted)
5   is
6     Local_Data : constant Integer := 100;
7
8     Extracted : Service_Extracted;
9
10    protected body Service_Extracted is
11      procedure Set is
12        begin
13          Extracted := True;
14        end Set;
15
16      function Get return Boolean is
17        (Extracted);
18      end Service_Extracted;
19
20      procedure Extract (Data : out Integer) is
21        Is_Extracted : constant Boolean := Extracted.Get;
22      begin
23        if not Is_Extracted then

```

(continues on next page)

(continued from previous page)

```
24     Data := Local_Data;
25     Extracted.Set;
26     else
27     Data := Integer'First;
28     end if;
29     end Extract;
30 end Service;
31
32 task body T is
33     X : Integer;
34     begin
35     loop
36     Service.Extract (X);
37     end loop;
38     end T;
39
40 end Example_10;
```

This code is correct. Abstract state is synchronized, hence can be accessed from multiple tasks and protected objects.

OBJECT-ORIENTED PROGRAMMING

5.1 What is Object Oriented Programming?

Object-oriented software construction is the building of software systems as structured collections of [...] abstract data type implementations.

Bertrand Meyer, “Object Oriented Software Construction”

- Object Oriented Programming is about:
 - isolating clients from implementation details (abstraction)
 - isolating clients from the choice of data types (dynamic dispatching)
- Object Oriented Programming is not:
 - the same as prototype programming (class and objects)
 - the same as scoping (class as the scope for methods)
 - the same as code reuse (use a component in a record in SPARK)

5.2 Prototypes and Scopes in SPARK

- Types in SPARK come with methods aka primitive operations

Listing 1: show_type_primitives.ads

```
1 package Show_Type_Primitives is
2
3   type Int is range 1 .. 10;
4   function Equal (Arg1, Arg2 : Int) return Boolean;
5   procedure Bump (Arg : in out Int);
6
7   type Approx_Int is new Int;
8   -- implicit definition of Equal and Bump for Approx_Int
9
10 end Show_Type_Primitives;
```

- Scope for the prototype is current declarative region
 - ... or up to the first freezing point – point at which the type must be fully defined, e.g. when defining an object of the type
- OOP without dynamic dispatching = Abstract Data Types

5.3 Classes in SPARK

- Classes in SPARK are tagged records

Listing 2: show_classes.ads

```
1 package Show_Classes is
2
3   type Int is tagged record
4     Min, Max, Value : Integer;
5   end record;
6
7   function Equal (Arg1, Arg2 : Int) return Boolean;
8   procedure Bump (Arg : in out Int);
9
10  type Approx_Int is new Int with record
11    Precision : Natural;
12  end record;
13  -- implicit definition of Equal and Bump for Approx_Int
14
15 end Show_Classes;
```

- Derived types are specializations of the root type
 - typically with more components
 - inheriting the methods on the parent type
 - can add their own methods

5.4 Methods in SPARK

- Derived methods can be overriding or not

Listing 3: show_derived_methods.ads

```
1 package Show_Derived_Methods is
2
3   pragma Elaborate_Body;
4
5   type Int is tagged record
6     Min, Max, Value : Integer := 0;
7   end record;
8
9   function Equal (Arg1, Arg2 : Int) return Boolean;
10  procedure Bump (Arg : in out Int);
11
12  type Approx_Int is new Int with record
13    Precision : Natural := 0;
14  end record;
15
16  overriding function Equal (Arg1, Arg2 : Approx_Int)
17    return Boolean;
18  overriding procedure Bump (Arg : in out Approx_Int);
19
20  not overriding procedure Blur (Arg : in out Approx_Int);
21
22 end Show_Derived_Methods;
```

Listing 4: show_derived_methods.adb

```

1 package body Show_Derived_Methods is
2
3   function Equal (Arg1, Arg2 : Int) return Boolean is
4     (Arg1 = Arg2);
5
6   procedure Bump (Arg : in out Int) is
7     Next : constant Integer := (if Arg.Value < Integer'Last
8                               then Arg.Value + 1
9                               else Integer'Last);
10
11   begin
12     if Next <= Arg.Max then
13       Arg.Value := Next;
14     end if;
15   end Bump;
16
17   overriding function Equal (Arg1, Arg2 : Approx_Int)
18     return Boolean is
19     (Arg1 = Arg2);
20
21   overriding procedure Bump (Arg : in out Approx_Int) is
22   begin
23     Bump (Int (Arg));
24   end Bump;
25
26   not overriding procedure Blur (Arg : in out Approx_Int) is
27     Prev : constant Integer := (if Arg.Value > Integer'First
28                               then Arg.Value - 1
29                               else Integer'First);
30
31   begin
32     if Arg.Value >= Prev then
33       Arg.Value := Prev;
34     end if;
35   end Blur;
36
37 end Show_Derived_Methods;

```

- Method called depends on static type

Listing 5: use_derived_methods.adb

```

1 with Show_Derived_Methods; use Show_Derived_Methods;
2
3 procedure Use_Derived_Methods is
4   I : Int;
5   AI : Approx_Int;
6 begin
7   Bump (I); -- call to Int.Bump
8   I.Bump; -- call to Int.Bump (object.method notation)
9
10  Bump (AI); -- call to Approx_Int.Bump
11  Bump (Int (AI)); -- call to Int.Bump
12 end Use_Derived_Methods;

```

5.5 Dynamic dispatching in SPARK

- Class-wide types
 - type of object that triggers dispatching
 - method called depends on dynamic type

Listing 6: use_dynamic_dispatching.adb

```

1 with Show_Derived_Methods; use Show_Derived_Methods;
2
3 procedure Use_Dynamic_Dispatching is
4
5   I : Int;
6   AI : Approx_Int;
7 begin
8   declare
9     IC : Int'Class := Int'Class (I);
10  begin
11    IC.Bump; -- call to Int.Bump
12  end;
13
14  declare
15    IC : Int'Class := Int'Class (AI);
16  begin
17    IC.Bump; -- call to Approx_Int.Bump
18  end;
19 end Use_Dynamic_Dispatching;

```

- Class-wide views of objects
 - in Ada, usually manipulated through pointers
 - in SPARK, manipulated through parameter passing

Listing 7: use_classwide_dispatching.adb

```

1 with Show_Derived_Methods; use Show_Derived_Methods;
2
3 procedure Use_Classwide_Dispatching is
4
5   procedure Call_Bump (Arg : in out Int'Class) is
6   begin
7     Arg.Bump;

```

(continues on next page)

(continued from previous page)

```

8   end Call_Bump;
9
10  I : Int;
11  AI : Approx_Int;
12
13  begin
14    Call_Bump (Int'Class (I)); -- calls Int.Bump(I)
15    Call_Bump (Int'Class (AI)); -- calls Approx_Int.Bump(AI)
16  end Use_Classwide_Dispatching;

```

5.5.1 A trivial example

- what is called here?

Listing 8: show_trivial_example.adb

```

1  procedure Show_Trivial_Example is
2
3    package Pkg_Trivial is
4      type Int is tagged record
5        Min, Max, Value : Integer;
6      end record;
7
8      procedure Bump (Arg : in out Int) is null;
9    end Pkg_Trivial;
10
11   use Pkg_Trivial;
12
13   procedure Call_Bump
14     (Arg : in out Int'Class) is
15   begin
16     Arg.Bump;
17   end Call_Bump;
18
19  begin
20    null;
21  end Show_Trivial_Example;

```

5.5.2 The problems with dynamic dispatching

- Control and data flow are not known statically
 - control flow - which subprogram is called when dispatching
 - data flow - what data this subprogram is accessing
 - similar to callbacks through subprogram pointers
- Avionics standard DO-178C lists 3 verification options
 - run all tests on parent type where derived type is used instead
 - cover all possible methods at dispatching calls
 - prove type substitutability (Liskov Substitution Principle aka LSP)

5.6 LSP - the SPARK solution to dynamic dispatching problems

- Class-wide contracts on methods
 - Pre'Class specifies strongest precondition for the hierarchy
 - Post'Class specifies weakest postcondition for the hierarchy

Listing 9: show_lsp.ads

```

1 package Show_LSP is
2
3   type Int is tagged record
4     Min, Max, Value : Integer := 0;
5   end record;
6
7   procedure Bump (Arg : in out Int) with
8     Pre'Class => Arg.Value < Arg.Max - 10,
9     Post'Class => Arg.Value > Arg.Value'Old;
10
11  type Approx_Int is new Int with record
12    Precision : Natural := 0;
13  end record;
14
15  overriding procedure Bump (Arg : in out Approx_Int) with
16    Pre'Class => Arg.Value > 100,
17    Post'Class => Arg.Value = Arg.Value'Old;
18
19 end Show_LSP;
```

Listing 10: show_lsp.ads

```

1 package Show_LSP is
2
3   type Int is tagged record
4     Min, Max, Value : Integer := 0;
5   end record;
6
7   procedure Bump (Arg : in out Int) with
8     Pre'Class => Arg.Value < Arg.Max - 10,
9     Post'Class => Arg.Value > Arg.Value'Old;
10
11  type Approx_Int is new Int with record
12    Precision : Natural := 0;
13  end record;
14
15  overriding procedure Bump (Arg : in out Approx_Int) with
16    Pre'Class => True,
17    Post'Class => Arg.Value = Arg.Value'Old + 10;
18
19 end Show_LSP;
```

Listing 11: show_lsp.ads

```

1 package Show_LSP is
2
3   type Int is tagged record
4     Min, Max, Value : Integer := 0;
5   end record;
6
```

(continues on next page)

(continued from previous page)

```

7  procedure Bump (Arg : in out Int) with
8     Pre'Class => Arg.Value < Arg.Max - 10,
9     Post'Class => Arg.Value > Arg.Value'Old;
10
11  type Approx_Int is new Int with record
12     Precision : Natural := 0;
13  end record;
14
15  overriding procedure Bump (Arg : in out Approx_Int);
16  -- inherited Pre'Class from Int.Bump
17  -- inherited Post'Class from Int.Bump
18
19  end Show_LSP;

```

5.6.1 Verification of dynamic dispatching calls

- Class-wide contracts used for dynamic dispatching calls

Listing 12: show_dynamic_dispatching_verification.adb

```

1  with Show_LSP; use Show_LSP;
2
3  procedure Show_Dynamic_Dispatching_Verification is
4
5     procedure Call_Bump (Arg : in out Int'Class) with
6         Pre => Arg.Value < Arg.Max - 10,
7         Post => Arg.Value > Arg.Value'Old
8     is
9     begin
10        Arg.Bump;
11    end Call_Bump;
12
13  begin
14    null;
15  end Show_Dynamic_Dispatching_Verification;

```

- LSP applies to data dependencies too
 - overriding method cannot read more global variables
 - overriding method cannot write more global variables
 - overriding method cannot have new input-output flows
 - SPARK RM defines Global'Class and Depends'Class (not yet implemented → use Global and Depends instead)

5.6.2 Class-wide contracts and data abstraction

- Abstraction can be used in class-wide contracts
- Typically use expression functions for abstraction

Listing 13: show_classwide_contracts.ads

```

1  package Show_Classwide_Contracts is
2
3     type Int is tagged private;
4

```

(continues on next page)

(continued from previous page)

```

5  function Get_Value (Arg : Int) return Integer;
6
7  function Small (Arg : Int) return Boolean with Ghost;
8
9  procedure Bump (Arg : in out Int) with
10     Pre'Class => Arg.Small,
11     Post'Class => Arg.Get_Value > Arg.Get_Value'Old;
12
13 private
14
15     type Int is tagged record
16         Min, Max, Value : Integer := 0;
17     end record;
18
19     function Get_Value (Arg : Int) return Integer is
20         (Arg.Value);
21     function Small (Arg : Int) return Boolean is
22         (Arg.Value < Arg.Max - 10);
23
24 end Show_Classwide_Contracts;

```

5.6.3 Class-wide contracts, data abstraction and overriding

- Abstraction functions can be overridden freely
 - overriding needs not be weaker or stronger than overridden

Listing 14: show_contract_override.ads

```

1  package Show_Contract_Override is
2
3     type Int is tagged record
4         Min, Max, Value : Integer := 0;
5     end record;
6
7     function Small (Arg : Int) return Boolean is
8         (Arg.Value < Arg.Max - 10);
9
10    type Approx_Int is new Int with record
11        Precision : Natural := 0;
12    end record;
13
14    overriding function Small (Arg : Approx_Int) return Boolean is
15        (True);
16
17 end Show_Contract_Override;

```

Listing 15: show_contract_override.ads

```

1  package Show_Contract_Override is
2
3     type Int is tagged record
4         Min, Max, Value : Integer := 0;
5     end record;
6
7     function Small (Arg : Int) return Boolean is
8         (Arg.Value < Arg.Max - 10);
9
10    type Approx_Int is new Int with record

```

(continues on next page)

(continued from previous page)

```

11     Precision : Natural := 0;
12 end record;
13
14 function Small (Arg : Approx_Int) return Boolean is
15     (Arg.Value in 1 .. 100);
16
17 end Show_Contract_Override;

```

- Inherited contract reinterpreted for derived class

Listing 16: show_contract_override.ads

```

1 package Show_Contract_Override is
2
3     type Int is tagged record
4         Min, Max, Value : Integer := 0;
5     end record;
6
7     procedure Bump (Arg : in out Int) with
8         Pre'Class => Arg.Value < Arg.Max - 10,
9         Post'Class => Arg.Value > Arg.Value'Old;
10
11     type Approx_Int is new Int with record
12         Precision : Natural := 0;
13     end record;
14
15     overriding procedure Bump (Arg : in out Approx_Int);
16     -- inherited Pre'Class uses Approx_Int.Small
17     -- inherited Post'Class uses Approx_Int.Get_Value
18
19 end Show_Contract_Override;

```

5.7 Dynamic semantics of class-wide contracts

- Class-wide precondition is the disjunction (or) of
 - own class-wide precondition, and
 - class-wide preconditions of all overridden methods
- Class-wide postcondition is the conjunction (and) of
 - own class-wide postcondition, and
 - class-wide postconditions of all overridden methods
- Plain Post + class-wide Pre / Post can be used together
- Proof guarantees no violation of contracts at runtime
 - LSP guarantees stronger than dynamic semantics

5.8 Redispatching and Extensions_Visible aspect

- Redispatching is dispatching after class-wide conversion
 - formal parameter cannot be converted to class-wide type when Extensions_Visible is **False**

Listing 17: show_redispatching.adb

```

1 with Show_Contract_Override; use Show_Contract_Override;
2
3 procedure Show_Redispatching is
4
5     procedure Re_Call_Bump (Arg : in out Int) is
6     begin
7         Int'Class (Arg).Bump;
8     end Re_Call_Bump;
9 begin
10    null;
11
12 end Show_Redispatching;
```

- Aspect Extensions_Visible allows class-wide conversion
 - parameter mode used also for hidden components

Listing 18: show_redispatching.adb

```

1 with Show_Contract_Override; use Show_Contract_Override;
2
3 procedure Show_Redispatching is
4
5     procedure Re_Call_Bump (Arg : in out Int)
6     with Extensions_Visible is
7     begin
8         Int'Class (Arg).Bump;
9     end Re_Call_Bump;
10 begin
11    null;
12
13 end Show_Redispatching;
```

5.9 Code Examples / Pitfalls

5.9.1 Example #1

Listing 19: oo_example_01.ads

```

1 package OO_Example_01 is
2
3     type Int is record
4         Min, Max, Value : Integer;
5     end record;
6
7     procedure Bump (Arg : in out Int) with
8         Pre'Class => Arg.Value < Arg.Max - 10,
9         Post'Class => Arg.Value > Arg.Value'Old;
```

(continues on next page)

(continued from previous page)

```

10
11 end OO_Example_01;

```

This code is not correct. Class-wide contracts are only allowed on tagged records.

5.9.2 Example #2

Listing 20: oo_example_02.ads

```

1 package OO_Example_02 is
2
3   type Int is tagged record
4     Min, Max, Value : Integer;
5   end record;
6
7   procedure Bump (Arg : in out Int) with
8     Pre => Arg.Value < Arg.Max - 10,
9     Post => Arg.Value > Arg.Value'Old;
10
11 end OO_Example_02;

```

This code is not correct. Plain precondition on dispatching subprogram is not allowed in SPARK. Otherwise it would have to be both weaker and stronger than the class-wide precondition (because they are both checked dynamically on both plain calls and dispatching calls).

Plain postcondition is allowed, and should be stronger than class-wide postcondition (plain postcondition used for plain calls).

5.9.3 Example #3

Listing 21: oo_example_03.ads

```

1 package OO_Example_03 is
2
3   pragma Elaborate_Body;
4
5   type Int is tagged record
6     Min, Max, Value : Integer;
7   end record;
8
9   procedure Bump (Arg : in out Int) with
10     Pre'Class => Arg.Value < Arg.Max - 10,
11     Post'Class => Arg.Value > Arg.Value'Old;
12
13   type Approx_Int is new Int with record
14     Precision : Natural := 0;
15   end record;
16
17   overriding procedure Bump (Arg : in out Approx_Int) with
18     Post'Class => Arg.Value = Arg.Value'Old + 10;
19
20 end OO_Example_03;

```

Listing 22: oo_example_03.adb

```

1 package body OO_Example_03 is
2
3   procedure Bump (Arg : in out Int) is
4     begin
5       Arg.Value := Arg.Value + 10;
6     end Bump;
7
8   overriding procedure Bump (Arg : in out Approx_Int) is
9     begin
10      Arg.Value := Arg.Value + 10;
11    end Bump;
12
13 end OO_Example_03;

```

This code is correct. Class-wide precondition of `Int.Bump` is inherited by `Approx_Int.Bump`. Class-wide postcondition of `Approx_Int.Bump` is stronger than the one of `Int.Bump`.

5.9.4 Example #4

Listing 23: oo_example_04.ads

```

1 package OO_Example_04 is
2
3   type Int is tagged record
4     Min, Max, Value : Integer;
5   end record;
6
7   function "+" (Arg1, Arg2 : Int) return Int with
8     Pre'Class => Arg1.Min = Arg2.Min
9     and Arg1.Max = Arg2.Max;
10
11  type Approx_Int is new Int with record
12    Precision : Natural;
13  end record;
14
15  -- inherited function "+"
16
17 end OO_Example_04;

```

This code is not correct. A type must be declared abstract or "+" overridden.

5.9.5 Example #5

Listing 24: oo_example_05.ads

```

1 package OO_Example_05 is
2
3   type Int is tagged record
4     Min, Max, Value : Integer;
5   end record;
6
7   procedure Reset (Arg : out Int);
8
9   type Approx_Int is new Int with record
10    Precision : Natural;

```

(continues on next page)

(continued from previous page)

```

11  end record;
12
13  -- inherited procedure Reset
14
15  end OO_Example_05;

```

This code is not correct. A type must be declared abstract or Reset overridden. Reset is subject to Extensions_Visible **False**.

5.9.6 Example #6

Listing 25: oo_example_06.ads

```

1  package OO_Example_06 is
2
3     type Int is tagged record
4         Min, Max, Value : Integer;
5     end record;
6
7     procedure Reset (Arg : out Int) with Extensions_Visible;
8
9     type Approx_Int is new Int with record
10         Precision : Natural;
11     end record;
12
13     -- inherited procedure Reset
14
15  end OO_Example_06;

```

Listing 26: oo_example_06.adb

```

1  package body OO_Example_06 is
2
3     procedure Reset (Arg : out Int) is
4     begin
5         Arg := Int'(Min => -100,
6                     Max  => 100,
7                     Value => 0);
8     end Reset;
9
10  end OO_Example_06;

```

This code is not correct. High: extension of Arg is not initialized in Reset.

5.9.7 Example #7

Listing 27: oo_example_07.ads

```

1  package OO_Example_07 is
2
3     pragma Elaborate_Body;
4
5     type Int is tagged record
6         Min, Max, Value : Integer;
7     end record;
8

```

(continues on next page)

(continued from previous page)

```

9   function Zero return Int;
10
11  procedure Reset (Arg : out Int) with Extensions_Visible;
12
13  type Approx_Int is new Int with record
14    Precision : Natural;
15  end record;
16
17  overriding function Zero return Approx_Int;
18
19  -- inherited procedure Reset
20
21  end OO_Example_07;
```

Listing 28: oo_example_07.adb

```

1  package body OO_Example_07 is
2
3    function Zero return Int is
4      ((0, 0, 0));
5
6    procedure Reset (Arg : out Int) is
7      begin
8        Int'Class (Arg) := Zero;
9      end Reset;
10
11   function Zero return Approx_Int is
12     ((0, 0, 0, 0));
13
14  end OO_Example_07;
```

This code is correct. Redispaching ensures that Arg is fully initialized on return.

5.9.8 Example #8

Listing 29: file_system.ads

```

1  package File_System is
2
3    type File is tagged private;
4
5    function Closed (F : File) return Boolean;
6    function Is_Open (F : File) return Boolean;
7
8    procedure Create (F : out File) with
9      Post'Class => F.Closed;
10
11   procedure Open_Read (F : in out File) with
12     Pre'Class  => F.Closed,
13     Post'Class => F.Is_Open;
14
15   procedure Close (F : in out File) with
16     Pre'Class  => F.Is_Open,
17     Post'Class => F.Closed;
18
19  private
20   type File is tagged record
21     Closed : Boolean := True;
22     Is_Open : Boolean := False;
```

(continues on next page)

(continued from previous page)

```

23  end record;
24
25  function Closed (F : File) return Boolean is
26    (F.Closed);
27
28  function Is_Open (F : File) return Boolean is
29    (F.Is_Open);
30
31  end File_System;

```

Listing 30: file_system.adb

```

1  package body File_System is
2
3    procedure Create (F : out File) is
4      begin
5        F.Closed := True;
6        F.Is_Open := False;
7      end Create;
8
9    procedure Open_Read (F : in out File) is
10     begin
11       F.Is_Open := True;
12     end Open_Read;
13
14    procedure Close (F : in out File) is
15     begin
16       F.Closed := True;
17     end Close;
18
19  end File_System;

```

Listing 31: oo_example_08.adb

```

1  with File_System; use File_System;
2
3  procedure OO_Example_08 is
4
5    procedure Use_File_System (F : out File'Class) is
6      begin
7        F.Create;
8        F.Open_Read;
9        F.Close;
10     end Use_File_System;
11
12  begin
13    null;
14  end OO_Example_08;

```

This code is correct. State automaton encoded in class-wide contracts is respected.

5.9.9 Example #9

Listing 32: file_system-sync.ads

```

1 package File_System.Sync is
2
3   type File is new File_System.File with private;
4
5   function Is_Synchronized (F : File) return Boolean;
6
7   procedure Create (F : out File) with
8     Post'Class => F.Closed;
9
10  procedure Open_Read (F : in out File) with
11    Pre'Class  => F.Closed,
12    Post'Class => F.Is_Open and F.Is_Synchronized;
13
14  procedure Close (F : in out File) with
15    Pre'Class  => F.Is_Open and F.Is_Synchronized,
16    Post'Class => F.Closed;
17
18 private
19   type File is new File_System.File with record
20     In_Synch : Boolean := True;
21   end record;
22
23   function Is_Synchronized (F : File) return Boolean is
24     (F.In_Synch);
25
26 end File_System.Sync;
```

Listing 33: file_system-sync.adb

```

1 package body File_System.Sync is
2
3   procedure Create (F : out File) is
4     begin
5       File_System.File (F).Create;
6       F.In_Synch := True;
7     end Create;
8
9   procedure Open_Read (F : in out File) is
10    begin
11      File_System.File (F).Open_Read;
12      F.In_Synch := True;
13    end Open_Read;
14
15   procedure Close (F : in out File) is
16    begin
17      File_System.File (F).Close;
18      F.Closed := True;
19    end Close;
20
21 end File_System.Sync;
```

Listing 34: oo_example_09.adb

```

1 with File_System.Sync; use File_System.Sync;
2
3 procedure OO_Example_09 is
4
```

(continues on next page)

(continued from previous page)

```

5  procedure Use_File_System (F : out File'Class) is
6  begin
7      F.Create;
8      F.Open_Read;
9      F.Close;
10 end Use_File_System;
11
12 begin
13     null;
14 end OO_Example_09;

```

This code is not correct. Medium: class-wide precondition might be stronger than overridden one

5.9.10 Example #10

Listing 35: file_system-sync.ads

```

1  package File_System.Sync is
2
3      type File is new File_System.File with private;
4
5      function Is_Synchronized (F : File) return Boolean;
6
7      procedure Create (F : out File) with
8          Post'Class => F.Closed;
9
10     procedure Open_Read (F : in out File) with
11         Pre'Class => F.Closed,
12         Post'Class => F.Is_Open;
13
14     procedure Close (F : in out File) with
15         Pre'Class => F.Is_Open,
16         Post'Class => F.Closed;
17
18     private
19     type File is new File_System.File with record
20         In_Synch : Boolean;
21     end record with
22         Predicate => File_System.File (File).Closed
23             or In_Synch;
24
25     function Is_Synchronized (F : File) return Boolean is
26         (F.In_Synch);
27
28 end File_System.Sync;

```

Listing 36: file_system-sync.adb

```

1  package body File_System.Sync is
2
3      procedure Create (F : out File) is
4      begin
5          File_System.File (F).Create;
6          F.In_Synch := True;
7      end Create;
8
9      procedure Open_Read (F : in out File) is

```

(continues on next page)

(continued from previous page)

```
10  begin
11      File_System.File (F).Open_Read;
12      F.In_Synch := True;
13  end Open_Read;
14
15  procedure Close (F : in out File) is
16  begin
17      File_System.File (F).Close;
18      F.Closed := True;
19  end Close;
20
21  end File_System.Sync;
```

Listing 37: oo_example_10.adb

```
1  with File_System.Sync; use File_System.Sync;
2
3  procedure OO_Example_10 is
4
5      procedure Use_File_System (F : out File'Class) is
6      begin
7          F.Create;
8          F.Open_Read;
9          F.Close;
10         end Use_File_System;
11
12     begin
13         null;
14     end OO_Example_10;
```

This code is correct. Predicate encodes the additional constraint on opened files. Type invariants are not yet supported on tagged types in SPARK.

GHOST CODE

6.1 What is ghost code?

ghost code is part of the program that is added for the purpose of specification

Why3 team, “The Spirit of Ghost Code”

... or verification

addition by SPARK team

- Examples of ghost code:
 - contracts (Pre, Post, Contract_Cases, etc.)
 - assertions (**pragma Assert**, loop (in)variants, etc.)
 - special values Func 'Result, Var 'Old, Var 'Loop_Entry
- Is it enough?

6.2 Ghost code - A trivial example

- how to express it?

Listing 1: show_trivial_example.ads

```
1 package Show_Trivial_Example is
2
3   type Data_Array is array (1 .. 10) of Integer;
4
5   Data : Data_Array;
6   Free : Natural;
7
8   procedure Alloc;
9
10 end Show_Trivial_Example;
```

Listing 2: show_trivial_example.adb

```
1 package body Show_Trivial_Example is
2
3   procedure Alloc is
4   begin
5     -- some computations here
6     --
7     -- assert that Free “increases”
8     null;
```

(continues on next page)

(continued from previous page)

```

9   end Alloc;
10
11  end Show_Trivial_Example;

```

6.3 Ghost variables - aka auxiliary variables

- Variables declared with aspect Ghost
 - declaration is discarded by compiler when ghost code ignored
- Ghost assignments to ghost variables
 - assignment is discarded by compiler when ghost code ignored

Listing 3: show_ghost_variable.ads

```

1  package Show_Ghost_Variable is
2
3     type Data_Array is array (1 .. 10) of Integer;
4
5     Data : Data_Array;
6     Free : Natural;
7
8     procedure Alloc;
9
10 end Show_Ghost_Variable;

```

Listing 4: show_ghost_variable.adb

```

1  package body Show_Ghost_Variable is
2
3     procedure Alloc is
4         Free_Init : Natural with Ghost;
5     begin
6         Free_Init := Free;
7         -- some computations here
8         pragma Assert (Free > Free_Init);
9     end Alloc;
10
11 end Show_Ghost_Variable;

```

6.4 Ghost variables - non-interference rules

- Ghost variable cannot be assigned to non-ghost one
 - Free := Free_Init;
- Ghost variable cannot indirectly influence assignment to non-ghost one

```

if Free_Init < Max then
  Free := Free + 1;
end if;

```

Listing 5: show_non_interference.adb

```

1  procedure Show_Non_Interference is
2
3     type Data_Array is array (1 .. 10) of Integer;
4
5     Data : Data_Array;
6     Free : Natural;
7
8     Free_Init : Natural with Ghost;
9
10    procedure Alloc is
11    begin
12        Free_Init := Free;
13        -- some computations here
14        pragma Assert (Free > Free_Init);
15    end Alloc;
16
17    procedure Assign (From : Natural; To : out Natural) is
18    begin
19        To := From;
20    end Assign;
21
22 begin
23     Assign (From => Free_Init, To => Free);
24 end Show_Non_Interference;

```

6.5 Ghost statements

- Ghost variables can only appear in ghost statements
 - assignments to ghost variables
 - assertions and contracts
 - calls to ghost procedures

Listing 6: show_ghost_statements.adb

```

1  procedure Show_Ghost_Statements is
2
3     type Data_Array is array (1 .. 10) of Integer;
4
5     Data : Data_Array;
6     Free : Natural;
7
8     Free_Init : Natural with Ghost;
9
10    procedure Alloc is
11    begin
12        Free_Init := Free;
13        -- some computations here
14        pragma Assert (Free > Free_Init);
15    end Alloc;
16
17    procedure Assign (From : Natural; To : out Natural)
18    with Ghost
19    is
20    begin

```

(continues on next page)

(continued from previous page)

```

21     To := From;
22     end Assign;
23
24 begin
25     Assign (From => Free, To => Free_Init);
26 end Show_Ghost_Statements;

```

```

procedure Show_Ghost_Statements is
begin
    -- Non-ghost variable "Free" cannot appear as actual in
    -- call to ghost procedure
    Assign (From => Free_Init, To => Free);
end Show_Ghost_Statements;

```

6.6 Ghost procedures

- Ghost procedures cannot write into non-ghost variables

```

procedure Assign (Value : Natural) with Ghost is
begin
    -- "Free" is a non-ghost variable
    Free := Value;
end Assign;

```

- Used to group statements on ghost variables
 - in particular statements not allowed in non-ghost procedures

```

procedure Assign_Cond (Value : Natural) with Ghost is
begin
    if Condition then
        Free_Init := Value;
    end if;
end Assign_Cond;

```

- Can have Global (including Proof_In) & Depends contracts

6.7 Ghost functions

- Functions for queries used only in contracts
- Typically implemented as expression functions
 - in private part - proof of client code can use expression
 - or in body - only proof of unit can use expression

Listing 7: show_ghost_function.ads

```

1 package Show_Ghost_Function is
2
3     type Data_Array is array (1 .. 10) of Integer;
4
5     Data : Data_Array;
6     Free : Natural;
7

```

(continues on next page)

(continued from previous page)

```

8   Free_Init : Natural with Ghost;
9
10  procedure Alloc with
11     Pre  => Free_Memory > 0,
12     Post => Free_Memory < Free_Memory'Old;
13
14  function Free_Memory return Natural with Ghost;
15
16  private
17
18     -- Completion of ghost function declaration
19     function Free_Memory return Natural is
20         (0); -- dummy implementation
21
22     -- If function body as declaration:
23     --
24     --     function Free_Memory return Natural is (...) with Ghost;
25
26  end Show_Ghost_Function;

```

6.8 Imported ghost functions

- Ghost functions without a body
 - cannot be executed

```
function Free_Memory return Natural with Ghost, Import;
```

- Typically used with abstract ghost private types
 - definition in SPARK_Mode(Off)
 - * type is abstract for GNATprove

Listing 8: show_imported_ghost_function.ads

```

1  package Show_Imported_Ghost_Function
2     with SPARK_Mode => On is
3
4     type Memory_Chunks is private;
5
6     function Free_Memory return Natural with Ghost;
7
8     function Free_Memory return Memory_Chunks
9         with Ghost, Import;
10
11  private
12     pragma SPARK_Mode (Off);
13
14     type Memory_Chunks is null record;
15
16  end Show_Imported_Ghost_Function;

```

- Definition of ghost types/functions given in proof
 - either in Why3 using External_Axiomatization
 - or in an interactive prover (Coq, Isabelle, etc.)

6.9 Ghost packages and ghost abstract state

- Every entity in a ghost package is ghost
 - local ghost package can group all ghost entities
 - library-level ghost package can be withed/used in regular units
- Ghost abstract state can only represent ghost variables

Listing 9: show_ghost_package.ads

```

1 package Show_Ghost_Package
2   with Abstract_State => (State with Ghost) is
3
4   function Free_Memory return Natural with Ghost;
5
6 end Show_Ghost_Package;
```

Listing 10: show_ghost_package.adb

```

1 package body Show_Ghost_Package
2   with Refined_State => (State => (Data, Free, Free_Init)) is
3
4   type Data_Array is array (1 .. 10) of Integer;
5
6   Data : Data_Array with Ghost;
7   Free : Natural with Ghost;
8
9   Free_Init : Natural with Ghost;
10
11  function Free_Memory return Natural is
12    (0); -- dummy implementation
13
14 end Show_Ghost_Package;
```

- Non-ghost abstract state can contain both ghost and non-ghost variables

6.10 Executing ghost code

- Ghost code can be enabled globally
 - using compilation switch -gnata (for all assertions)
- Ghost code can be enabled selectively
 - using `pragma Assertion_Policy (Ghost => Check)`
 - SPARK rules enforce consistency - in particular no write disabled

Listing 11: show_exec_ghost_code.ads

```

1 package Show_Exec_Ghost_Code is
2
3   pragma Assertion_Policy (Ghost => Check);
4   -- pragma Assertion_Policy (Ghost => Ignore, Pre => Check);
5
6   procedure Alloc with
7     Pre => Free_Memory > 0;
8
9   function Free_Memory return Natural with Ghost;
```

(continues on next page)

(continued from previous page)

```

10
11 end Show_Exec_Ghost_Code;

```

- GNATprove analyzes all ghost code and assertions

6.11 Examples of use

6.11.1 Encoding a state automaton

- Tetris in SPARK
 - at [Tetris³](#)
- Global state encoded in global ghost variable
 - updated at the end of procedures of the API

```

type State is (Piece_Falling, ...) with Ghost;
Cur_State : State with Ghost;

```

- Properties encoded in ghost functions

```

function Valid_Configuration return Boolean is
  (case Cur_State is
    when Piece_Falling => ...,
    when ...)
with Ghost;

```

6.11.2 Expressing useful lemmas

- GCD in SPARK
 - at [GCD⁴](#)
- Lemmas expressed as ghost procedures

```

procedure Lemma_Not_Divisor (Arg1, Arg2 : Positive) with
  Ghost,
  Global => null,
  Pre  => Arg1 in Arg2 / 2 + 1 .. Arg2 - 1,
  Post => not Divides (Arg1, Arg2);

```

- Most complex lemmas further refined into other lemmas
 - code in procedure body used to guide proof (e.g. for induction)


³ <http://blog.adacore.com/tetris-in-spark-on-arm-cortex-m4>

⁴ <http://www.spark-2014.org/entries/detail/gnatprove-tips-and-tricks-proving-the-ghost-common-denominator-gcd>

6.11.3 Specifying an API through a model


- Red black trees in SPARK
 - at [Red black trees](#)⁵
- Invariants of data structures expressed as ghost functions
 - using `Type_Invariant` on private types
- Model of data structures expressed as ghost functions
 - called from `Pre / Post` of subprograms from the API
- Lemmas expressed as ghost procedures
 - sometimes without contracts to benefit from inlining in proof

6.12 Extreme proving with ghost code - red black trees in SPARK



⁵ <http://www.spark-2014.org/entries/detail/research-corner-auto-active-verification-in-spark>

6.13 Positioning ghost code in proof techniques



6.14 Code Examples / Pitfalls

6.14.1 Example #1

Listing 12: example_01.adb

```

1  procedure Example_01 is
2
3     type Data_Array is array (1 .. 10) of Integer;
4
5     Data : Data_Array;
6     Free : Natural;
7
8     procedure Alloc is
9         Free_Init : Natural with Ghost;
10    begin
11        Free_Init := Free;
12        -- some computations here
13        if Free <= Free_Init then
14            raise Program_Error;
15        end if;
16    end Alloc;
17 begin
18     null;
19
20 end Example_01;

```

This code is not correct. A ghost entity cannot appear in this context.

6.14.2 Example #2

Listing 13: example_02.adb

```
1 procedure Example_02 is
2
3   type Data_Array is array (1 .. 10) of Integer;
4
5   Data : Data_Array;
6   Free : Natural;
7
8   procedure Alloc is
9     Free_Init : Natural with Ghost;
10
11     procedure Check with Ghost is
12       begin
13         if Free <= Free_Init then
14           raise Program_Error;
15         end if;
16       end Check;
17     begin
18       Free_Init := Free;
19       -- some computations here
20       Check;
21     end Alloc;
22 begin
23   null;
24
25 end Example_02;
```

This code is correct. Note that procedure Check is inlined for proof (no contract).

6.14.3 Example #3

Listing 14: example_03.ads

```

1 package Example_03 is
2
3   type Data_Array is array (1 .. 10) of Integer;
4
5   Data : Data_Array;
6   Free : Natural;
7
8   pragma Assertion_Policy (Pre => Check);
9
10  procedure Alloc with
11     Pre => Free_Memory > 0;
12
13  function Free_Memory return Natural with Ghost;
14
15 end Example_03;

```

This code is not correct. Incompatible ghost policies in effect during compilation, as ghost code is ignored by default. Note that GNATprove accepts this code as it enables all ghost code and assertions.

6.14.4 Example #4

Listing 15: example_04.ads

```

1 package Example_04 is
2
3   procedure Alloc with
4     Post => Free_Memory < Free_Memory'Old;
5
6   function Free_Memory return Natural with Ghost;
7
8 end Example_04;

```

Listing 16: example_04.adb

```

1 package body Example_04 is
2
3   Free : Natural;
4
5   Max : constant := 1000;
6
7   function Free_Memory return Natural is
8   begin
9     return Max - Free + 1;
10  end Free_Memory;
11
12  procedure Alloc is
13  begin
14    Free := Free + 10;
15  end Alloc;
16
17 end Example_04;

```

This code is not correct. No postcondition on Free_Memory that would allow proving the postcondition on Alloc.

6.14.5 Example #5

Listing 17: example_05.ads

```

1 package Example_05 is
2
3   procedure Alloc with
4     Post => Free_Memory < Free_Memory'Old;
5
6   function Free_Memory return Natural with Ghost;
7
8 end Example_05;
```

Listing 18: example_05.adb

```

1 package body Example_05 is
2
3   Free : Natural;
4
5   Max : constant := 1000;
6
7   function Free_Memory return Natural is (Max - Free + 1);
8
9   procedure Alloc is
10  begin
11    Free := Free + 10;
12  end Alloc;
13
14 end Example_05;
```

This code is correct. `Free_Memory` has an implicit postcondition as an expression function.

6.14.6 Example #6

Listing 19: example_06.adb

```

1 procedure Example_06 is
2
3   subtype Resource is Natural range 0 .. 1000;
4   subtype Num is Natural range 0 .. 6;
5   subtype Index is Num range 1 .. 6;
6   type Data is array (Index) of Resource;
7
8   function Sum (D : Data; To : Num) return Natural is
9     (if To = 0 then 0 else D (To) + Sum (D, To - 1))
10    with Ghost;
11
12   procedure Create (D : out Data) with
13     Post => Sum (D, D'Last) < 42
14   is
15   begin
16     for J in D'Range loop
17       D (J) := J;
18       pragma Loop_Invariant (2 * Sum (D, J) <= J * (J + 1));
19     end loop;
20   end Create;
21
22 begin
23   null;
24 end Example_06;
```

This code is not correct. Info: expression function body not available for proof (Sum may not return).

6.14.7 Example #7

Listing 20: example_07.adb

```

1  procedure Example_07 is
2
3     subtype Resource is Natural range 0 .. 1000;
4     subtype Num is Natural range 0 .. 6;
5     subtype Index is Num range 1 .. 6;
6     type Data is array (Index) of Resource;
7
8     function Sum (D : Data; To : Num) return Natural is
9         (if To = 0 then 0 else D (To) + Sum (D, To - 1))
10        with Ghost, Annotate => (GNATprove, Terminating);
11
12    procedure Create (D : out Data) with
13        Post => Sum (D, D'Last) < 42
14    is
15    begin
16        for J in D'Range loop
17            D (J) := J;
18            pragma Loop_Invariant (2 * Sum (D, J) <= J * (J + 1));
19        end loop;
20    end Create;
21
22    begin
23        null;
24    end Example_07;

```

This code is correct. Note that GNATprove does not prove the termination of Sum here.

6.14.8 Example #8

Listing 21: example_08.adb

```

1  procedure Example_08 is
2
3     subtype Resource is Natural range 0 .. 1000;
4     subtype Num is Natural range 0 .. 6;
5     subtype Index is Num range 1 .. 6;
6     type Data is array (Index) of Resource;
7
8     function Sum (D : Data; To : Num) return Natural is
9         (if To = 0 then 0 else D (To) + Sum (D, To - 1))
10        with Ghost, Annotate => (GNATprove, Terminating);
11
12    procedure Create (D : out Data) with
13        Post => Sum (D, D'Last) < 42
14    is
15    begin
16        for J in D'Range loop
17            D (J) := J;
18        end loop;
19    end Create;
20

```

(continues on next page)

(continued from previous page)

```

21 begin
22     null;
23 end Example_08;

```

This code is correct. The loop is unrolled by GNATprove here, as D'Range is 0 .. 6. The automatic prover unrolls the recursive definition of Sum.

6.14.9 Example #9

Listing 22: example_09.adb

```

1 with Ada.Containers.Functional_Vectors;
2
3 procedure Example_09 is
4
5     subtype Resource is Natural range 0 .. 1000;
6     subtype Index is Natural range 1 .. 42;
7
8     package Seqs is new
9         Ada.Containers.Functional_Vectors (Index, Resource);
10    use Seqs;
11
12    function Create return Sequence with
13        Post => (for all K in 1 .. Last (Create'Result) =>
14                Get (Create'Result, K) = K)
15    is
16        S : Sequence;
17    begin
18        for K in 1 .. 42 loop
19            S := Add (S, K);
20        end loop;
21        return S;
22    end Create;
23
24 begin
25     null;
26 end Example_09;

```

This code is not correct. Loop requires a loop invariant to prove the postcondition.

6.14.10 Example #10

Listing 23: example_10.adb

```

1 with Ada.Containers.Functional_Vectors;
2
3 procedure Example_10 is
4
5     subtype Resource is Natural range 0 .. 1000;
6     subtype Index is Natural range 1 .. 42;
7
8     package Seqs is new
9         Ada.Containers.Functional_Vectors (Index, Resource);
10    use Seqs;
11
12    function Create return Sequence with
13        Post => (for all K in 1 .. Last (Create'Result) =>

```

(continues on next page)

(continued from previous page)

```
14         Get (Create'Result, K) = K)
15     is
16     S : Sequence;
17     begin
18         for K in 1 .. 42 loop
19             S := Add (S, K);
20             pragma Loop_Invariant (Integer (Length (S)) = K);
21             pragma Loop_Invariant
22                 (for all J in 1 .. K => Get (S, J) = J);
23         end loop;
24         return S;
25     end Create;
26
27 begin
28     null;
29 end Example_10;
```

This code is correct.

TEST AND PROOF

7.1 Various Combinations of Tests and Proofs

- Overall context is functional verification of code
- Combination can take various forms:
 - Test before Proof – contracts used first in test, possibly later in proof
 - Test for Proof – contracts executed in test to help with development of proof
 - Test alongside Proof – some modules are tested and other modules are proved
 - Test as Proof – exhaustive test as good as proof
 - Test on top of Proof – proof at unit level completed with test at integration level, also using contracts

7.2 Test (be)for(e) Proof

7.2.1 Activating Run-time Checks

- Need to activate run-time checks in executable
- Constraint_Error exceptions activated by default
 - Use `-gnat-p` to revert effect of previous `-gnatp` (say in project file)
 - Use `-gnato` to activate overflow checking (default since GNAT 7.3)
- Special handling of floating-point computations
 - Use `-gnateF` to activate bound checking on standard float types
 - Use `-msse2 -mfpmath=sse` to forbid use of 80bits registers and FMA on x86 processors
 - Runtime/BSP should enforce use of Round-Nearest-tie-to-Even (RNE) rounding mode

7.2.2 Activating Assertions

- Need to activate assertions in executable
- `Assertion_Error` exceptions deactivated by default
 - Use `-gnata` to activate globally
 - Use `pragma Assertion_Policy` to activate file-by-file
 - Use `-gnateE` to get more precise error messages (`Contract_Cases`)
- Special assertions checked at run time
 - `Contract_Cases` → checks one and only one case activated
 - `Loop_Invariant` → checks assertion holds (even if not inductive)
 - `Assume` → checks assertion holds (even if not subject to proof)
 - `Loop_Variant` → checks variant decreases wrt previous iteration

7.2.3 Activating Ghost Code

- Need to activate ghost code in executable
- Ghost code, like assertions, is deactivated by default
 - Use `-gnata` to activate globally
 - Use `pragma Assertion_Policy (Ghost => Check)` to activate locally
- Inconsistent combinations will be rejected by GNAT
 - Ignored ghost entity in activated assertion
 - Ignored ghost assignment to activated ghost variable

7.3 Test for Proof

7.3.1 Overflow Checking Mode

- Problem: ignore overflow checks in assertions/contracts
 - Only applies to signed integer arithmetic
 - Does not apply inside an expression function returning an integer
- Solution: use unbounded arithmetic in assertions/contracts
 - Will use 64bits signed arithmetic when sufficient
 - Otherwise use a run-time library for unbounded arithmetic
- Two ways to activate unbounded arithmetic
 - Use `-gnato13` compiler switch
 - Use `pragma Overflow_Mode` with arguments (`General => Strict, Assertions => Eliminated`) in configuration pragma file

7.4 Test alongside Proof

7.4.1 Checking Proof Assumptions

- Need to check dynamically the assumptions done in proof
 - Postcondition of tested subprogram called in proved subprogram
 - Precondition of proved subprogram called in tested subprogram
- Other assumptions beyond pre- and postconditions
 - Global variables read and written by tested subprogram
 - Non-aliasing of inputs and outputs of proved subprogram
 - No run-time errors in tested subprogram
- GNATprove can list assumptions used in proof
 - Switch `--assumptions` adds info in `gnatprove.out` file
- See "Explicit Assumptions - A Prenup for Marrying Static and Dynamic Program Verification"

7.4.2 Rules for Defining the Boundary

- `SPARK_Mode` defines a simple boundary test vs. proof
 - Subprograms with `SPARK_Mode (On)` should be proved
 - Subprograms with `SPARK_Mode (Off)` should be tested
- `SPARK_Mode` can be used at different levels
 - Project-wise switch in configuration pragma file (with value `On`) → explicit exemptions of units/subprograms in the code
 - Distinct GNAT project with `SPARK_Mode (On)` for proof on subset of units
 - Explicit `SPARK_Mode (On)` on units that should be proved
- Unproved checks inside proved subprograms are justified
 - Use of pragma `Annotate` inside the code

7.4.3 Special Compilation Switches

- Validity checking for reads of uninitialized data
 - Compilation switch `-gnatVa` enables validity checking
 - pragma `Initialize_Scalars` uses invalid default values
 - Compilation switch `-gnateV` enables validity checking for composite types (records, arrays) → extra checks to detect violation of SPARK stronger data initialization policy
- Non-aliasing checks for parameters
 - Compilation switch `-gnateA` enables non-aliasing checks between parameters
 - Does not apply to aliasing between parameters and globals

7.5 Test as Proof

7.5.1 Feasibility of Exhaustive Testing

- Exhaustive testing covers all possible input values
 - Typically possible for numerical computations involving few values
 - e.g. OK for 32 bits values, not for 64 bits ones
 - * binary op on 16 bits → 1 second with 4GHz
 - * unary op on 32 bits → 1 second with 4GHz
 - * binary op on 32 bits → 2 years with 64 cores at 4GHz
 - In practice, this can be feasible for trigonometric functions on 32 bits floats
- Representative/boundary values may be enough
 - Partitioning of the input state in equivalent classes
 - Relies on continuous/linear behavior inside a partition

7.6 Test on top of Proof

7.6.1 Combining Unit Proof and Integration Test

- Unit Proof of AoRTE combined with Integration Test
 - Combination used by Altran UK on several projects
 - Unit Proof assumes subprogram contracts
 - Integration Test verifies subprogram contracts
- Unit Proof of Contracts combined with Integration Test
 - Test exercises the assumptions made in proof
 - One way to show Property Preservation between Source Code and Executable Object Code from DO-178C/DO-333
 - * Integration Test performed twice: once with contracts to show they are verified in EOC, once without to show final executable behaves the same

7.7 Test Examples / Pitfalls

7.7.1 Example #1

I am stuck with an unproved assertion. My options are:

- switch --level to 4 and --timeout to 360
- open a ticket on GNAT Tracker
- justify the unproved check manually

Evaluation: This approach is not correct. Why not, but only after checking this last option:

- run tests to see if the assertion actually holds

7.7.2 Example #2

The same contracts are useful for test and for proof, so it's useful to develop them for test initially.

Evaluation: This approach is not correct. In fact, proof requires more contracts than test, as each subprogram is analyzed separately. But these are a superset of the contracts used for test.

7.7.3 Example #3

Assertions need to be activated explicitly at compilation for getting the corresponding run-time checks.

Evaluation: This approach is correct. Use switch `-gnata` to activate assertions.

7.7.4 Example #4

When assertions are activated, loop invariants are checked to be inductive on specific executions.

Evaluation: This approach is not correct. Loop invariants are checked dynamically exactly like assertions. The inductive property is not something that can be tested.

7.7.5 Example #5

Procedure P which is proved calls function T which is tested. To make sure the assumptions used in the proof of P are verified, we should check dynamically the precondition of T.

Evaluation: This approach is not correct. The precondition is proved at the call site of T in P. But we should check dynamically the postcondition of T.

7.7.6 Example #6

Function T which is tested calls procedure P which is proved. To make sure the assumptions used in the proof of P are verified, we should check dynamically the precondition of P.

Evaluation: This approach is correct. The proof of P depends on its precondition being satisfied at every call.

7.7.7 Example #7

However procedure P (proved) and function T (tested) call each other, we can verify the assumptions of proof by checking dynamically all preconditions and postconditions during tests of T.

Evaluation: This approach is not correct. That covers only functional contracts. There are other assumptions made in proof, related to initialization, effects and non-aliasing.

7.7.8 Example #8

Proof is superior to test in every aspect.

Evaluation: This approach is not correct. Maybe for the aspects Pre and Post. But not in other aspects of verification: non-functional verification (memory footprint, execution time), match with hardware, integration in environment... And testing can even be exhaustive sometimes!

7.7.9 Example #9

When mixing test and proof at different levels, proof should be done at unit level and test at integration level.

Evaluation: This approach is not correct. This is only one possibility that has been used in practice. The opposite could be envisioned: test low-level functionalities (e.g. crypto in hardware), and prove correct integration of low-level functionalities.

7.7.10 Example #10

There are many ways to mix test and proof, and yours may not be in these slides.

Evaluation: This approach is correct. YES! (and show me yours)